
AVERTISSEMENT

La présente brochure contient en 3 éme partie un complément au manu-
el MEM/DOS pour 1'APPLE //c.

Cette 3 éme partie ne décrit que les différences apparaissant pour
1'utilisateur entre la version Apple //e et 1la version Apple //c.

Le lecteur qui posséde un Apple //c devra donc lire d'abord dans la
partie Apple //c le chapitre concernant la prise en main du systéme,
et effectuer le démarrage sur son Apple //c, puis lire et exécuter
les exemples du manuel d'initiation a MEM/DOS.

Lorsqu'il sera familiarisé avec le systéme, le lecteur pourra alors
se référer au manuel d'utilisation sur 1'Apple //e, et se reporter
pour certains points précis a la 3 éme partie lorsque des différences
avec 1'Apple //c seront signalées dans la marge.

Introduction
QUELS SONT LES OBJECTIFS ET LES AVANTAGES DE MEM/DOQS ?

MEM/DOS est un PRECIEUX OUTIL DE DEVELOPPEMENT pour les programmeurs
réalisant des logiciels d'application de gestion en basic. Il peut :

. gérer facilement des fichiers & accés par clé,

. saisir des données par masques d'écran,

. structurer des programmes avec des sous-programmes,

. manipuler sans erreur d'arrondi des nombres de 12, 20, 48 chiffres,

. imprimer des états par masques d'impression,

. travailler rapidement gréce a la simplicité de la syntaxe et I'aide d’utilitaires de dévelop-
pement nombreux et puissants.

MEM ¥ DOS est aussi un puissant SYSTEME D’EXPLOITATION de gestion capable de
gérer des disques et des disquettes (Disk Operating System). Les programmeurs systéme,
réalisant des configurations micro-informatique, peuvent :

. accéder rapidement et facilement aux informations du disque,

. bénéficier d'une gestion dynamique de disquettes ou disques durs pouvant aller de 140K
a 120 Mégaoctets,

. modifier et adapter, grace a des utilitaires puissants, le systéme lui-méme en fonction de
son environnement et de leurs propres besoins.

MEM / DOS est pour I'entreprise qui s’équipe d'un systéme micro-informatique I'assu-
rance d'un LOGICIEL DE QUALITE : les fonctionnalités du systeme d’exploitation,
apportent a I'utilisateur confort, rapidité et sécurité.

QUELS SONT LES BUTS DE CETTE NOTICE ?

On ne trouvera dans cette notice que les rudiments nécessaires a I'utilisation de
MEM / DOS pour les petites applications de gestion. Sa lecture suppose une connais-
sance de BASIC et quelques rudiments du DOS.

Aprés avoir consacré quelques heures a la lecture de cette notice, il sera possible de
passer au manuel d utilisation.

. Le premier chapitre permet.de prendre contact avec MEM/DOS et de gérer des program-.
mes. Le deuxiéme est consacré aux fichiers. le troisieme aux masques et le quatrieme aux
sous-programmes. Enfin le chapitre V illustre par un exemple un peu plusimportant I"utili-
sation conjointe des fichiers, des masques et des sous-programmes.

Pour des raisons matérielles, les programmes illustrant cette notice sont rassemblés a la fin
des chapitres ot ils sont présentés ; ces programmes, qui sont numérotés de PO aP12 et de
SPQ & SP5. se trouvent sur la disquette. Il est préférable de les étudier et de les utiliser dans
I'ordre.

Dans ce chapitre, vous verrez d’abord comment mettre en route MEM/DOS, ce qui
est vraiment facile ; vous ferez ensuite une espéce de survol des notions nécessaires a son
utilisation, aussi bien du point de vue de la syntaxe des commandes que de la définition des
“objets” manipulés : fichiers et masques. Enfin, vous apprendrez & gérer sous MEM/DOS
les programmes écrits en BASIC : vous constaterez qu'il n'y a pas une grande différence
avec les commandes habituelles du DOS 3.3 mais qu'il y a déja quelques possibilités
supplémentaires.

1. DEMARRAGE DE MEM/DOS

1.1 Implantation physique

Pour utiliser MEM/DOS, il nous faut :

- un apple Il 48 K ou un Applell e

- un lecteur de disquette avec contréleur et le DOS 3.3
- une carte MEM/DOS

La carte MEM/DOS peut étre fixée sur n'importe quel support (*'slot”) de votre Apple.
Cependant, en cas d' utilisation d’une carte Z 80, la carte MEM/DOS doit étre placée sur
un support de numéro supérieur a celui de la carte Z 80.

Enfin, on suppose que votre contrdleur de disque est sur le support numéro 6.

1.2 Ce qu’il faut faire une fois pour toutes

La disquette fournie par MEMSOFT ne contient pas le DOS 3.3. Il faut donc, et ceci, une
fois pour toutes, “installer” le DOS 3.3 sur votre disquette MEM/DOS. Comme toujours,
il vous est fortement conseillé d’effectuer une copie de la disquette MEM/DOS avant
toute manipulation. Voici donc la séquence d'opérations a effectuer :

- Faire une copie de la disquette MEM/DOS.

Cette disquette n’est absolument pas protégée et n'importe quel utilitaire de copie
conviendra (COPYA du DOS 3.3. par exemple).

- Votre Apple étant sous DOS 3.3, mettre la disquette MEM/DOS et taper
RUN HELLO
Ce programme HELLO permet également de transcoder d'anciens fichiers DOS 3.3 en

vue de leur utilisation sous MEM/DOS, mais ce n'est pas ce qui nous intéresse en ce
moment. C’est pourquoi on choisit I'option 1 proposée par ce programme.




- Vous assistez alors & un défilement de plusieurs écrans (dont un écran en 80 colonnes,
si vous avez une carte 80) colonnes sur votre Apple). Le menu de démarrage ci-dessous
s affiche enfin :

08 6502
FRANCE
1982

MEM/
MIs
({}

D
)

B BRSIC @ corlE DE DISRUE
GESTION DE MASQUES [l CATALOGUE

GESTION DE FICHIER [(J FORMATAGE DISOUE
DEMO MASOUE El CuPIE AUTOMATIGUE
DEMO FICHIER TESTS

VOTRE CHOIX ?

Vous étes maintenant en possession d'une disquette MEM / DOS qui démarrera toute
seule ; d'ot...

1.3 Ce qu’il suffit de faire les autres fois

Le démarrage normal du MEM / DOS sur Apple s'effectuera désormais de la fagon
suivante : introduire la disquette MEM/DOS (qui contient maintenant le DOS 3.3) dans
le tourne-disquette et mettre votre Apple sous tension. Le systéme se charge lui-méme
et le message suivant apparait a I'écran quelques instants :

Un écran de 80 colonnes apparait éventuellement (si vous avez une carte 80 colonnes,
bien sir), puis le menu de démarrage s'affiche, et MEMDOS est a votre disposition.

1.4 Le menu de démarrage

Les dix fonctions proposées par ce menu sont détaillées en Annexe Ill mais leur compré-
hension ne semblera aisée au débutant qu’aprés avoir pris connaissance du contenu de
cette brochure.

Pour l'instant, nous allons utiliser le Basic et il faut donc taper 1. SurI'écran qui s'affiche,
le curseur est au début du mot RUN écrit sur la derniére ligne. Il vous suffit de taper
RETURN pour faire disparaitre ce RUN et passer 4 la ligne suivante. Vous étes mainte-

nant en Basic mais... sous MEM/DOS : vous pouvez constater que vous n’étes plus sous
DOS 3.3 : par exemple CATALOG ne marche pas !!!

Remarque : Si, par erreur, vous tapez RETURN au lieu d'un chifre, le menu de démar-
rage, vous pouvez revenir a ce menu en tapant : RUN puis RETURN.




2 . LES DIFFERENTS OBJETS QUE L'ON TROUVE SUR UNE DISQUETTE

Le travail d'un systeme d'exploitation est de gérer une disquette, c’ est-a-dire de permettre
al'utilisateur d'y écrire et d'y lire desinforrhations. Avant de voir quelles sont les commandes
qui permettent de gérer effectivement ces informations, commencons par voir quels sont
les ** objets " {objet = groupement d'informations) que I'on peut gérer.

2 . 1 Avant toute chose : la numérotation des lecteurs

Dans le DOS normal, les lecteurs sont repérés par $6,D1, puis S6,D2 ... De méme, dans le
systeme PASCAL, ces lecteurs sont repérés par # 4: puis #5: ... lls s'appellent A: et B:

sous CP/M ! On voit donc que chaque systéme baptise a sa guise les lecteurs. Dans le cas
du MEM/DOS::

le lecteur connecté en $6,D1 porte le numéro 0
le lecteur connecté en S6,D2 porte le numéro 1
le lecteur connecté en S5,D1 porte le numéro 2



2 - 2 Le catalogue d’une disquette

La commande LET“*” a pour effet d’afficher le catalogue du lecteur courant ; on entend
par lecteur courant celui que I'on a utilisé en dernier : par exemple, aprés avoir démarré le
systéme, le lecteur courant est évidemment le lecteur numéro 0 (S6, D1), mais aprés avoir
exécuté une commande telle que :

LET”*1”

qui agit sur le lecteur 1, le lecteur courant est maintenant le lecteur 1 et la commande
LET”*”

agira sur le lecteur 1 (ce n’est pas la seule commande a agir sur le lecteur courant).

Bien entendu, avec MEM/DOS qui ne dispose que du mini-lecteur numeéroté 0,
inutile de préciser que le lecteur courant est toujours 0 et qu’on ne doit pas le modifier : si
vous tapez LET*1”  vous ferez apparaitre un message d’erreur, ce qui est logique
puisqu’il n’y a pas de périphérique correspondant, mais cela va encore plus loin : pour
remettre les choses en ordre, il faut taper LET“* 0 car le lecteur courant était devenu le
lecteur 1!

2 - 3 Les différents types d’objets
Revenons au catalogue : nous voyons que les identificateurs sont précédés par une des
lettres B, F, G, M ou P qui signifient respectivement que I'objet en question est :
B : un Bloc de valeurs Binaires recopiées telles quelles,
: un Fichier de données,
: un Programme écrit en BASIC

: un Masque (pas d’affolement, nous verrons bientét ce que c’est)

O 2 v

: un Groupe de masques (appelé parfois masque Global .

2 - 4 Le nom des objets

Des objets peuvent avoir le méme nom s'ils sont de type différent. Le nom d’un objet est

toujours composé de 20 caractéres ; si I'utilisateur en donne moins MEM/DOS complete

par des blancs. On peut utiliser tous les caractéres et les blancs sont pris en compte. Donc :
" SALAIRES BRUTS

est un nom d'objet différent de

SALAIRES BRUTS




2 - 5 Le catalogue partiel

L utilisateur perspicace aura constaté que le catalogue présente d’abord les objets du type
B, puis ceux du type F, etc ... et que les objets du méme type sont classés par ordre lexico-
graphique (ordre alphabétique stendu a tous les caractéres : FACT 1 sera avant FACT 2).
On peut restreindre I'affichage du catalogue aux objets d’un seul type.

LET™* 0,PROGRAM”

par e.xemple, si on veut uniquement la liste des programmes du lecteur O.

3 . PRINCIPES D’UTILISATION DES FICHIERS ET DES MASQUES

Le butdece paragraphe estsimplement d’ exposer a ceux des lecteurs qui sont“débutants”
des dafinitions bien connues des informaticiens.

3 - 1 Qu’est-ce qu’un fichier de données ?

Signalons d’abord que “‘données” est ici une traduction de I'anglais DATA ... qui signifie
“information”. Donc, fichier de données doit se comprendre fichier d’informations : que
ces informations soient données ou calculées ne change rien ! Il s'agit d’une de ces ambi-
guités génantes mais bien ancrées dans le vocabulaire des informaticiens.

Apres cette précision, voyons le terme “fichier” : avant I'avénement de I'informatique une
fiche était (d’apres le Robert) “un carton sur lequel on inscrit des renseignements en vue
d’un classement” et le mot fichier avait le sens courant d’ensemble de fiches ou de boite
contenant un ensemble de fiches. En remplacant “carton” par “emplacement sur un
support magnétique” on a la notion de fiche et de fichier informatique (bizarrement les
informaticiens utilisent le mot enregistrement de préférence au mot fiche).

A Torigine les fichiers étaient sur bande magnétique et, les fiches étant 'une derriére
I'autre, on était obligé de les parcourir toutes, séquentiellement, si on en cherchait une en
particulier. Actuellement, les fichiers sont sur disquettes et on peut accéder plus facilement
a n’importe quel enregistrementpuisque la disquette est““découpée” en morceaux appelés
“secteurs” que I'on peut atteindre directement. Il n’empéche qu’on utilise encore des mots
comme ‘“séquentiel”, “accés direct”, “relatif” pour exprimer des modes de
parcours d’un fichier ; on verra en détail dans le chapitre Il consacré aux fichiers que
MEM/DOS est particulierement agréable et efficace dans ce domaine.




3 - 2 Qu’est-ce qu’un masque ?

La saisie des données sous mode “conversationnel” classique s’effectue sous forme d’un
dialogue entre le programme qui affiche sur I'écran un message précisant ce qu'il attend, et
I'utilisateur qui tape en retour au clavier la valeur attendue, Cela va bien tant qu’il y a peu
de données a saisir, mais lorsqu'il s'agit de saisir les données relatives a plusieurs enregis-
trements d’un fichier, ce processus peut paraitre désagréable a I'usager et on a imaginé de
faire demander en bloc I'ensemble des ditférentes variables d'un enregistrement (par
exemple : nom, prénom, numéro de sécurité sociale, date de naissance) grace a une
technique nouvelle : le masque.

Un ensemble de messages et de “fenétres” est affiché en bloc sur I'écran : dans chaque
fenétre I'utilisateur entre la valeur de la variable demandée et, lorsque toutes les fenétres
sont remplies, une commande permet d’envoyer I'ensemble des valeurs au programme.
Le chapitre Ill donne tous les détails sur I utilisation de masques en MEM/DOS.

3 - 3 La notion d’identificateur temporaire

Tout fichier ou masque est répertorié dans le catalogue d’une disquette grace a son nom
(ou identificateur). Mais lorsqu’on veut utiliser dans un programme un de ces objets, on le
fait par I'intermédiaire d’un identificateur temporaire (appelé classiquement “numeéro
logique”): dans un méme programme, chaque objet utilisé se voit d’abord attribuer un
identificateur temporaire différent qui, par la suite, sera utilisé a la place de son nom.

Cet identificateur temporaire est en fait formé d’un seul caractére... mais quelconque :
chiffre, lettre, signe (a I'exception de $). La commande

LET” # OPEN:B,FICHIER:ADRESSE”

attribue I'identificateur temporaire B au fichier ADRESSE et, dans la suite de ce programme,
la commande

LET”READ:B”

déclenchera une lecture sur le fichier ADRESSE. Ne vous préoccupez pas pour I'instant du
détail de ces commandes, nous y reviendrons plus tard.

4 - GESTION DES PROGRAMMES BASIC

Avant d’entamer les chapitres suivants consacrés aux masques et aux fichiers, voyons
d’abord comment MEM/DOS gare les programmes écrits en BASIC : c’est voisin de ce
que vous avez I'habitude de faire. Il y a simplement une syntaxe légérement différente et
quelques possibilités supplémentaires.

4 - 1 Les messages d’erreurs

A partir de maintenant, vous allez probablement multiplier les petits essais au clavier... ce
qui entrainera quelques erreurs. Alors, autant consacrer un paragraphe a ce sujet avant de
poursuivre... Il existe deux classes d’erreurs : celles qui n’interrompent pas le déroulernent
du programme, et que 'on qualifie de “récupérables” et les autres, qui sont baptisées “irré-
cupérables ou fatales”, en ce sens qu'elles interrompent le déroulement normal du
programme.

Les erreurs irrécupérables sont des erreurs de programmation (donc détectées a l'interpré-
tation ou a I'exécution), ou des erreurs “systémes” : disque plein ou protégé en écriture par
exemple. Les erreurs récupérables n'interrompent pas le déroulement du programme ; le
programmeur peut constater comment la commande s’est exécutée : la variable WS est
réservée par MEM / DOS pour y placer un “compte-rendu de transaction” : si une
commande s'est bien exécutée, MEM/DOS place la valeur 0 dans cette variable ; en cas de
difficults, MEM/DOS vy place une valeur positive. Un bon programmeur fera donc suivre
toute commande MEM/DOS d’un test lui permettant de vérifier si WS est devenu
positif et prendra des mesures en conséquence.

Remarques :

R.1: Laliste des messages d’erreurs et des valeurs correspondantes de WS pour les erreurs
récupérables est donnée en annexe l.

R.2 : Lorsqu’une commande MEM/DOS est exécutée en mode direct (ou interactif : c’est-
a-dire lorsque, hors de tout programme, une commande est entrée au clavier) et que
I'exécution de cette commande provoque une erreur, le message DIRECT ERROR apparait.
Ce message est parfois mal interprété : il ne veut pas dire que la commande en question est
interdite en mode direct, il signifie simplement que I'exécution de cette commande n’a pu
étre menée a bien.

R.3 : Ce message est de plus suivi d'un “message explicite” qui évite a I'utilisateur de faire
imprimer WS pour voir ce qui n'a pas bien marché.

R.4 : N'oubliez pas que WS est un identificateur de variable réservé au systeme. Ne I'utili-
sez donc surtout pas dans vos programmes.

4 - 2 Sauvegarde d’un programme.

Aprés avoir tapé un petit programme du genre

NEW
10 PRINT "OK”
20 END

~11 -




Il est facile de le sauvegarder sur la disquette courante sous un nom quelconque, SE par
exemple, grace a la commande :

SAVE"SE™
Si on veut préciser le numéro du lecteur, c’est possible :

SAVE"0:SE”

On peut alors constater par la commande LET “*” que SE apparait dans la liste des
programmes du catalogue.

4 - 3 Effacement

L'opération qui permet de supprimer un objet de la disquette est déclenchée par la
commande DELETE qui veut que I'on précise le type de I'objet & supprimer, puisque des
objets de type différent peuvent avoir le méme nom :

LET” # DELETE,PROGRAM:SE” ou LET“# D,P: SE”

aura pour effet d’effacer le programme SE du disque courant. Le mot PROGRAM peut
étre résumé a son initiale P, et DELETE & D.

4 - 4 Sauvegarde aprés effacement

La commande SAVE "SE” ne marche pas s'il y a déja un programme appelé SE sur la
disquette : cela permet aux étourdis de ne pas effacer par mégarde un programme dont ils
avaient oublié le nom ! Mais il existe une variante de la commande SAVE qui permet
d’effacer volontairement une version précédente :

SAVE” @ SE” ou SAVE” @ 0:SE” (notez bien la position du @ )
4 - 5 Chargement d’un programme

La commande LOAD "SE” a pour effet d’effacer le programme brécédemment en
mémoire centrale et d'y placer le programme SE.

4 - 6 Chargement en queue

La commande LOAD” # SE™" a pour effet de placer le texte du programme SE a la suite du
texte résidant en mémoire centrale : ¢’est trés utile pour mettre bout a bout des morceaux
de programmes ... & condition d’avoir prévu la numérotation adéquate car ce chargement
bout a bout s’effectue sans s’ occuper des numeéros de lignes : il n’y a pas fusion des deux
programmes !!! Par curiosité, vous pouvez exécuter la séquence

NEW

LOAD "SE”
LOAD " #SE”
LIST

Si vous étes vraiment curieux, enlevez le premier END et exécutez !!!

4 - 7 Exécution d’'un programme

La commande RUN lance I'exécution de ce qui se trouve en mémoire centrale. La
commande RUN "TOTO” cherche le programme TOTO sur la mini-disquette, le charge
en mémoire centrale et I'exécute. Par contre, si vous essayez d’exécuter un masque ou un
fichier, cela ne marchera pas ; on verra aux chapitres Il et Ill comment on se sert de ces
objets.

4 - 8 Exécution avec conservation de variables pré-valuées

Lorsqu’on lance I'exécution d’'un programme par un RUN, toutes les variables sont
remises a zéro. Par contre, la commande

LOAD’/ESSAI”

a pour effet de conserver les valeurs des variables actuellement en mémoire centrale et de
lancer I'exécution avec ces valeurs aprés avoir chargé ESSAI

La séquence

NEW

10 PRINT "OK”
20 PRINTM

30 END

SAVE “ESSAI”
M=10

LOAD "VESSAI”

affiche OK puis 10 sur I’écran.

REMARQUE : 1l s’agit bien du mot LOAD et pas du mot RUN. Attention ceci ne marche
que si ESSAI est d'une taille inférieure au dernier programme chargé normalement.

5 - LE POINT SUR LA FORME DES COMMANDES MEM/DOS.

Il'y a deux types de commandes sous MEM/DOS. D’une part celles qui agissent sur un
programme (SAVE, LOAD, RUN) que nous venons d’utiliser, et d’autre part... toutes les
autres : nous avons utilise LET”*” nous avons vu sans les détailler LET”"READ,B” et
LET” # OPEN:B,FICHIER, ADRESSE” et nous en verrons beaucoup d’autres bient6t.

Vous avez pu constater que la syntaxe des premieres est trés voisine de ce que vous avez
I'habitude d’ utiliser : la différence essentielle est la présence de guillemets ; toutes les autres
utilisent le mot LET. C’est “le” mot réservé du MEM / DOS, en ce sens qu'il intervient
souvent pour donner une commande au systéme ; il n’a plus du tout la signification du
BASIC normal otiil indiquait, de maniére facultative et peu employée, I'instruction d’affec-
tation (LET N = N + 1 par exemple). LET est suivi d’une expression chaine précisant ce
qu’il faut faire ; cette chaine peut contenir des constantes et des variables ; par exemple, si
Kestla van'able (numeérique) dont la valeur est le numéro du lecteur, on aurait pu écrire :

LET"*,”+STR$(K)

REMARQUES :

R.1: Le caractére , est un séparateur. Il en existe trois autres, qui sont - ; : tous les quatre
sont interchangeables entre eux : dans une commande MEM/DOS vous pouvez utiliser a
votre guise I'un ou 'autre.

-, ; séparateurs interchangeables dans une commande MEM/DOS.

R.2: Les blancs n’ont pas de signification : ils sont ignorés par le systéme, sauf dans le nom
d’un objet comme on I'a vu plus haut.

R.3: Les motsréservés READ, OPEN, etc... peuvent étre abrégés a leur premiére lettre : de
toutes facons, MEM/DOS n’analyse que la premiére lettre de chaque mot réservé. Cette
remarque ne s’applique pas aux identificateurs de commande SAVE, LOAD, RUN, LET
qui eux sont analysés par BASIC. Autrement dit, tout ce qui suit est équivalent :

LET” # OPEN-1,FICHIER:FSE”
LET”# O-1,F:FSE”
LET”# OUVRIRLE-1,FICHIER:FSE”

R.4 : Les commandes SAVE, LOAD, RUN sont aussi suivies d’une chaine de caracteéres :
c’est ce qui explique la présence des guillemets ; cela autorise I'usage de variables dans les
commandes.

6 - UN PETIT PROBLEME NON RESOLU

Tout ordre de MEM /DOS suivant le THEN doit étre précédé d’un : sous peine d’une
erreur fatale se traduisant par un message SYNTAX ERROR. 1l faut donc écrire par
exemple :

IFA=BTHEN:LET "...

Clest tres irritant quand on tombe dans cette chausse-trappe ... alors essayez de retenir
cette petite imperfection de MEM/DOS.

7 - UN DILEMME PEDAGOGIQUE

Y a-t-il eu d’abord une poule ou un ceuf ? Autrement dit : est-ce I’ceuf qui vient de la poule
ou la poule de I'ceuf ? Le dilemme provient de I'indissociabilité des deux protagonistes.

De la méme facon, il est difficile de décider s’il vaut mieux expliquer d’abord Iutilisation
des fichiers ou celle des masques puisque malheureusement I'exposé de la premiére
notion se fera sans I'aide de la deuxieme. Nous avons choisi d’exposer d’abord I’ utilisation
des fichiers (chapitre II) ; aprées la lecture du chapitre [l consacré aux masques, vous
pourrez facilement reprendre les exemples donnés au chapitre Il en vous servant de
masques. Vous constaterez alors la pleine puissance de MEM/DOS.

Chapitre Il :

UTILISATION DES FICHIERS

1 L’ORGANISATION DES FICHIERS

1 - 1 Description du contenu d’une fiche

1 - 2 Classement des fiches entre elles : organisation du fichier
1 - 3 Rangement logique et rangement physique

1 - 4 Fichier MEM/DOS = structure + fiches

2 LOGIQUE DE L'UTILISATION DES FICHIERS

2 - 1 Actions élémentaires au niveau de la fiche, du fichier
2 - 2 Organisation des traitements faisant intervenir des fichiers

3 LA SYNTAXE DES COMMANDES
4 LA COMMANDE CLEAR
5 UN EXEMPLE SIMPLE

6 LES PRINCIPALES COMMANDES

6 - 1 Description de la fiche et du type de fichier

6 - 2 Commandes au niveau du fichier

6 - 3 Commandes agissant sur les enregistrements d’un fichier relatif

6 - 4 Commandes agissant sur les enregistrements d’un fichier indexé

8 LES PROGRAMMES P0 A P9

Ce chapitre commence par donner quelques précisions sur I'organisation des fichiers et les
principes logiques de leur utilisation. Ces précisions ne sont évidemment que des rappels
pour certains lecteurs qui pourront donc survoler les paragraphes 1 et 2. La suite du chapi-
tre est consacrée a I'utilisation des fichiers en MEM/DOS et elle s’appuie sur quelques
petits programmes rassemblés sous forme de fiches au paragraphe 8 ; les programmes sont
également présents sur la mini-disquette MEM/DOS. Vous étes fortement invité a
faire tourner ces programmes le moment venu.

1 L’ORGANISATION DES FICHIERS

Sil'on veut s’y retrouver dans une collection d’objets quelconques, il faut un peu d’organi-
sation. Il en va de méme pour les fichiers informatisés, d’autant plus que les *“fiches” sont ici
illisibles a I'ceil nu et qu’en cas de manque d’organisatian, I' utilisateur sera fortement péna-
lisé. Il faut donc parfaitement définir ce que contient une fiche, comment on les classe entre
elles et comment on utilise le fichier.

1 - 1 Description du contenu d’une fiche

Une fiche (ou enregistrement) est composée d'une ou plusieurs rubriques (on dit aussi
caractéristiques ou champs) qui sont repérées chacune par un identificateur de variables.
MEM/DOS autorise les variables classiques du BASIC : entier, flottant, chaine de caracte-
res, mais aussi deux autres types de variables : les variables représentant une date qui
seront de type DATE et les variables représentant un entier compris entre 0 et 255 qui
seront de type ENTIER COURT ; I'utilisation des variables de type DATE ou ENTIER
COURT est exposée en annexe Il : ne mélangeons pas tout.

Dans une fiche, on peut également mémoriser des tableaux de toutes ces variables, mais
nous n’en parlons pas dans cette notice d’initiation.

1 - 2 Classement des fiches entre elles : organisation du fichier

Il'y a deux méthodes possibles, a priori, pour repérer les fiches entre elles :

- Chaque fois qu’on ajoute une fiche dans le fichier, le systéme lui donne le numéro N + 1
et c’est ce numéro qui la repére : les fiches sont ordonnées (classées) selon ce numéro
auquel il faudra associer une variable pour pouvoir le manipuler. Cette variable est parfois
appelée “pointeur du fichier”.

- On peut par contre étre amené a traiter les fiches dans un ordre lié & la valeur d’une
variable : les fiches seront alors repérées grace a cette rubrique que I'on appellera “clé du
fichier” ou “index”.

Un fichier sera qualifié de relatif si ses fiches sont rangées dans I'ordre chronologique de
création et sera qualifié d’indexé si ses fiches sont rangées selon une clé ; on verra plus loin
que la clé peut en fait étre constituée de plusieurs rubriques ... et méme qu'un fichier peut
avoir plusieurs clés différentes.

1-3 Rangement logique et rangement physique

On vient de voir que, dans un fichier relatif, I'ordre “logique” de rangement des fiches était
I'ordre de création. Sur la disquette, les fiches sont effectivement rangées dans cet ordre.
On dit que le rangement physique coincide avec le rangement logique. Cela peut surpren-
dre, mais dans un fichier indexé, une nouvelle fiche est également placée physiquement en
bout de fichier quelle que soit la valeur de la clé !!!

L’explication est que MEM/DOS gére en fait un “fichier d’index’” associé & tout fichier
indexé : c’est une espéce de tableau ou les fiches (repérées par leur adresse sur le disque)
sont rangées selon la valeur de la clé : cette technique permet de n’avoir pas a déplacer
physiquement toute une partie du fichier chaque fois qu'’il faut insérer une nouvelle fiche,
ce qui risquerait d’étre long.

Remarques :

R.1: Deux ou plusieurs fiches peuvent avoir la méme valeur de clé ; elles sont alors rangées
“logiquement” I'une derriére 'autre.

R.2 : Mais I utilisateur peut demander 8 MEM/DOS de ne pas accepter de nouvelle fiche
correspondant a une valeur de la clé qui existe déja dans le fichier : il y a une commande
spéciale pour cela (cf paragraphe 5).

R.3: [l est recommandé de réorganiser de temps en temps le fichier pour que I'ordre logi-
que coincide avec I'ordre physique, ce qui diminue le temps d’accés : cela se fait grice a la
commande REORGANISE.

R.4 : Rien n’interdit d’associer plusieurs fichiers d'index a un méme fichierindexé : c’est ce
qui permet d’avoir plusieurs clés pour un méme fichier.

R.5: ATintention du lecteur soucieux de précision, on peut signaler que MEM/DOS range
tous les fichiers en utilisant la technique des B-arbres. Cette technique est utilisée par [BM
dans tous ses gros ordinateurs sous le nom de VSAM. Le meilleur article a ce sujet est
malheureusement en anglais : “The Ubiquitous B-Tree” par D. COMER dans Computing
Surveys Vol. 11, numeéro 2, Juin 1979 pages 121 a 137.

1 - 4 Fichier MEM/DOS = Structure + fiches

Dans la plupart des systemes d’exploitation des micros ordinateurs, un fichier ne contient
aucun renseignement concernant sa propre nature ou la structure de ses fiches ; ceci est
une source inépuisable d’erreurs. Par contre, un fichier MEM/DOS mémorise deux types
de renseignements :

- un ensemble d’informations, parfois appelé “bibliothéque”, concernant le fichier lui-
méme : type du fichier, nom du pointeur ou nom et type des variables clés, description de
la fiche : nom et type des variables qui la composent. Cette bibliotheque est accessible a
I'utilisateur qui peut la visualiser & tout moment.

- un ensemble de 0 & N fiches, c’est-a-dire de valeurs affectées aux diverses variables de
chaque fiche.

Cette organisation permet 8 MEM/DOS de vérifier que I'utilisation d’un fichier est bien
conforme a la structure qui lui a &té donnée lors de la création et donc d’éviter certaines
erreurs d’inattention.

2 LOGIQUE DE L’UTILISATION DES FICHIERS

2 - 1 Actions élementaires au niveau de la fiche, du fichier

Une fiche peut étre créée, remplie, lue, modifiée ou supprimée. Cela va de soi et ne néces-
site pas d’autres explications ; par contre, en ce qui concerne les actions au niveau d’un
fichier, on va voir (a I'intention des débutants en informatique) & quoi correspondent les
opérations de définition, création, ouverture et fermeture d’un fichier.

2 - 2 Organisation des traitements faisant intervenir des fichiers

Pour pouvoir utiliser un fichier, il y a une chose a faire “une fois pour toutes”, c’est de
préciser au systeme :

- la description de la fiche : ¢’est-a-dire la liste de ses rubriques,
- le type d’organisation du fichier : relatif ou indexé, et le nom de I'index ou du pointeur,
- le nom sous lequel ce fichier sera mémorisé sur le disque.

Ces précisions sont données au systéme lors de la création du fichier ; en MEM/DOS, la
création se fait en deux étapes, donc en deux commandes : 'une de description, I'autre de
création proprement dite. La syntaxe de ces commandes est donnée plus loin.

On a déja vu que dans tout programme utilisant un fichier, il faut lui attribuer un identifica-
teur temporaire dont on se sert ensuite plutét que d’ utiliser le nom sous lequel ce fichier est
mémoriseé sur disque. Cette précision est donnée lors d’une phase appelée ouverture qui a
pour effets secondaires d’abord de préparer le systéme a travailler sur la premiére fiche
mais aussi de réserver quelques octets en mémoire centrale dans lesquels transiteront les
informations : ces octets forment ce qu’on appelle un tampon (ou buffer).

Lorsqu'un programme cesse d’utiliser un fichier, on peut récupérer la place mémoire
réservée a ce fichier grace a une opération appelée classiquement fermeture du fichier, par
opposition a I'ouverture. Précisons a ce sujet qu’avec MEM/DOS toute opération d’écri-
ture dans un fichier sur disque est exécutée sur le champ, contrairement a ce qui se passe
dans la plupart des systemes oti la recopie du buffer est différée aussi longtemps que possi-
ble. L’ opération de “fermeture” correspond donc a une simple récupération de la place du
buffer, c’est pourquoi la commande correspondante est appelée CLEAR plutét que
CLOSE. Le gros avantage qu’apporte MEM / DOS avec cette technique de recopie
immeédiate, c’est une sécurité accrue : on ne peut pas perdre une écriture disque (ou
plusieurs) comme cela arrive avec les autres systémes en cas de “pépin”’ ;avec MEM/DOS,
un fichier est modifié dés qu’une instruction d’écriture est exécutée ... ce qui est normal
apres tout ! '

Remarques :

R.1 : La création d’un fichier entraine implicitement son ouverture. Donc si, dans un
méme programme, vous créez un fichier et I'utilisez immédiatement, il ne faut pas essayer
de I'ouvrir aprés I'avoir créé, cela entrainerait une erreur.

R.2 : Les identificateurs temporaires d’un fichier donné ne sont pas forcément les mémes
dans les différents programmes qui I’ utilisent.

3 LA SYNTAXE DES COMMANDES

Une commande fichier devra permettre 8 MEM/DOS de savoir sur quel fichier il doit tra-
vailler : grace & I'identificateur temporaire ; puis de déterminer s'il s’agit d’'une commande
au niveau du fichier ou d’une fiche : le caractére # indiquera une action au niveau fichier ;
enfin, certaines commandes étant identiques pour les fichiers, les programmes et les mas-
ques, il faudra préciser le type d’objet en question.

LET ”# OPEN-1,FICHIER.FSE ”
FSE : nom du fichier sur
le disque courant
FICHIER : action sur un fichier
1 : identificateur temporaire
OPEN : ouverture d’un objet
# : action au niveau fichier

Rappels :
LET”#0:1;F-FSE”

aura le méme effet puisque seule la premiére lettre des mots réservés est analysée et que
les séparateurs : ; , - sont équivalents. ‘

50 PRINT "DONNEZ UN NOM DE FICHIER”
60 INPUT NF$
70 LET "# O:1; F-"+NF$

aura le méme effet sur le fichier dont le nom sera entré au clavier : on ouvrira avec I'iden-
tificateur temporaire 1 dans le programme le fichier répertorié sur le disque courant sous le
nom lu au clavier.

Si vous essayez maintenant (ou plus tard bien sur !) cette commande en mode direct et que
le fichier FSE n’existe pas sur le disque courant, vous aurez un message DIRECT ERROR.
Rappelons que cela ne veut pas dire que cette commande est interdite en mode direct.

4 LA COMMANDE CLEAR

Nous consacrons un paragraphe spécial & la commande CLEAR car sa syntaxe est simple,
mais son utilisation est un peu difficile & expliquer. La commande

LET”# CLEAR:3”

a pour effet de rompre le lien entre 'identificateur temporaire 3 et le fichier qu’il identifiait
jusqu’alors : accessoirement cela libere les quelques octets du buffer en plus de I'identifica-
teur 3 que I'on peut allouer & un autre objet. Par assimilation avec un terme couramment
employé en informatique, on appellera FERMETURE du fichier 3 cette opération. La com-
mande

LET”# CLEAR,$”

a pour effet de “fermer” tous les fichiers et aussi tous les masques (une fermeture qui n’est
donc qu’une libération d’un identificateur temporaire et de quelques octets a le méme effet
sur un masque). Cette instruction de fermeture générale est bien agréable car elle évite de
répéter autant de CLEAR qu'’il y a d’objets a fermer.

Jusque la, normalement, c’est clair. Ce qui pose probléme, c’est de savoir ol mettre une
telle instruction de fermeture générale. A premiére vue, on est tenté de la mettre en fin de
programme ; mais si, pour une raison quelconque, I'exécution du programme ne se
termine pas normalement, les identificateurs temporaires ne sont pas libérés et si un autre
programme essaie d’en réutiliser un immédiatement, on est certain de voir apparaitre un
message d’erreur. La seconde possibilité est donc de mettre cette instruction de fermeture
générale en téte de chaque programme plutét qu’en queue : c’est une solution un peu
égoiste du genre “aprés moi le déluge...”. Alors la bonne solution, c’est peut-étre d’en
mettre une en téte par prudence, et une en queue par politesse !

5 UN EXEMPLE SIMPLE

Avant d’entrer dans les détails de I'utilisation des fichiers, regardez (aprés avoir lu les
remarques ci-dessous) ce petit exemple concernant un fichier relatif : dans un premier
programme PO on crée un fichier (page 25) et le programme P1 permet de remplir des
fiches, les programmes P2, P3 et P4 montrent deux méthodes différentes de lecture dans
ce fichier (pages 27, 28, 29).

Avant que vous ne partiez a la découverte de ces programmes (qui sont rassemblés sous
forme de fiches (!) a la fin de ce chapitre), il faut rappeler qu’au chapitre III, on verra I utili-
sation des masques qui permettent de gérer agréablement les entrées-sorties : les exem-
ples que vous allez voir sont la pour vous permettre de débuter et de bien comprendre le
détail de certains mécanismes. Regardez les fiches de ces cinq programmes sans oublier de
lire les remarques AVANT de faire exécuter les programmes correspondants, mais sans
chercher & comprendre en détail la syntaxe des instructions, cela sera vu dans quelques

pages.

6 LES PRINCIPALES COMMANDES

Apreés 'étude de ces cinq petits programmes nous pouvons maintenant regarder de
maniére un peu plus précise (mais toutefois pas dans tous les détails) les différentes
commandes.

6 - 1 Description de la fiche et du type de fichier

Dans une seule et méme commande, on va décrire la composition de la fiche, c’est-a-dire
la liste de ses rubriques, et le type de fichier (relatif ou indexé) ainsi que le nom de la (ou
des) variable(s) qui commande(nt) I'accés.

Fichier relatif :
LET™ @ TA% = T%,XR$”

C’est le symbole > qui indique 8 MEM/DOS que cette commande lui décrit un fichier.
Ensuite, le caractere @ indique qu'il s’agit d'un fichier relatif. Enfin, le caractére = sépare
la variable entigre qui précise le pointeur ou numéro (TA%), de toutes celles qui précisent
les rubriques (T%, XR$).

Fichier indexé par une clé composée d’une seule variable :

Premiere différence avec ce qui précede : la disparition du symbole @ . La variable clé
peut étre de type quelconque : entier, réel ou chaine ; si la variable clé est une chaine de
caracteres, il faut préciser la longueur maximale qu'aura cette chaine : |

LET"»NM$12=N%,SC,AD$”

Notez bien la syntaxe : lalongueur n’est pas entre parenthéses. Notez également que siune
rubrique est une chaine, il n'y a pas a préciser la longueur de cette rubrique : seule la
longueur de la chaine clé est & préciser : les enregistrements ont une longueur variable, ce
qui économise de la place sur le disque.

Fichier indexé par une clé composée de plusieurs variables :
LET"»NM$10,PR$15AN=N%,SC,AD$”

Le caractere = sépare la liste des variables de la clé de laliste des variables de la fiche. Les
fiches seront classées selon la premiére variable de la clé et, en cas d’égalité, selon la
deuxieme, ... etc : si NM$ est un nom et PR$ un prénom, le fichier sera classé dans I'ordre
habituel. En cas d’homonymie parfaite, et si AN est 'année de naissance, I'ainé précédera
le cadet. Rappellons que vous ne devez préciser la longueur d'une variable de la clé que si
c’est une chaine.

Fichier indexé par plusieurs clés :
Les clés sont séparées par le symbole & ; chaque clé peut étre composée de plusieurs
variables :

LET”)NM$10,PR$20&SS$13=NM$,PR$,SS$,T%,XR$”
On constate que les variables composant les clés sont répétées dans la description de la
fiche : en effet, qu'il y ait une ou plusieurs clés, les variables qui composent laoulesclés ne
font pas partie de la fiche ; donc, si on accéde a une fiche gracea lavaleur d’une clé et si on

lit cette fiche, on n'y trouvera la valeur des autres clés de cette fiche... que si on lesya
répétées comme rubriques.

Bien entendu, lors de I'utilisation d’un tel fichier, il faudra préciser quelle clé on veut utiliser:
si un fichier correspondant & cette description s’est vu attribuer I'identificateur temporaire
4, la commande LET "READ-4,2” déclenchera une lecture par I'intermédiaire de la
deuxiéme clé (SS$).

6 - 2 Commandes au niveau du fichier :

Certaines commandes sont identiques, quelle que soit le type du fichier :
LET ”# NEW-3 FICHIER:FSE”

crée et ouvre sous l'identificateur temporaire 3 le fichier qui s’appellera FSE sur la disquette ;
la description de ce fichier doit avoir été précisée auparavant.

LET ”# OPEN-2,FICHIER:0:FSE”
ouvre le fichier FSE de la disquette 0 sous I'identificateur temporaire 2
LET ”# DELETE, FICHIER:FSE”

supprime le fichier FSE de la disquette. Ici, il n'y a pas besoin de passer par un identificateur
temporaire. On ne peut pas supprimer un fichier ouvert.

On peut enfin demander 8 MEM/DOS de nous révéler quelle est la description d’un fichier
existant : la séquence

10 LET” # OPEN-1,FICHIER:FSE”
20 LET”ENTER-1”
30 LET"VISUALISE”

affiche la liste de la ou les clés et des rubriques du fichier FSE. Attention, cette séquence est
inutilisable en mode direct.

REMARQUE : Toutes ces commandes ont comme premier caractéere ~ qui symbolise une
action au niveau d’un “objet” ; on va voir inmédiatement que les commandes ayant une
action au niveau d’une fiche ne comportent pas ce caractére

6 - 3 Commandes agissant sur les enregistrements d’un fichier relatif :

On a déja vu certaines de ces commandes :
LET "READ-2”

lit ' enregistrement (du fichier de I'identificateur temporaire 2) dont le numéro a été préala-
blement précisé.

LET”NEXT-1"

lit P enregistrement “suivant” : aprés I'ouverture, c’est le premier enregistrement qui est lu ;
s'il n’y a pas d’enregistrement suivant, la variable WS regoit la valeur 255, alors qu’elle
recoit 0 s’il n’y a pas de probléme de lecture.

LET "WRITE-3"

ajoute un enregistrement a la queue du fichier : I'utilisateur n'a pas 3 se préoccuper de la
gestion du numéro de fiche : MEM/DOS s'en occupe.

1l existe d'autres commandes lout aussi utiles :
LET "UPDATE-2"

remplace les valeurs des variables de I'enregistrement sur la disquetle par les valeurs
qu'ont ces variables en mémoire centrale.

LET "DELETE-4"

efface un enregistrement ; attention : les numéros des fiches restantes ne sont pas modifiés:
il se crée un “trou” dans la séquence des numéros des enregistrements et ce irou ne pourra
plus étre rempli !

6 - 4 Commandes aglssant sur les enregistrements d’'un fichler indexé:

Les cing commandes READ, NEXT, WRITE, UPDATE, DELETE agissent également dans
le cas d'un fichier indexé avec la m@me syntaxe, mals il y a quelques différences dans les
conditions d’emplol ou les conséquences :

- READ, UPDATE, DELETE exigent que la valeur de la clé solt préalablement fixée,

- DELETE supprime blen un enregistrement, mais la valeur de la clé est réutilisable ulté-
rieurement,

- WRITE suppose que I'utilisateur a précisé les valeurs de toutes les clés s'lly en aplusleurs.

Enfin, il existe une commande ADD qui joue un rdle analogue a celul de WRITE A ceci prés

ue MEM/DOS acceptera les “homonymes" éventuels : on pourra avolr dans le fichier
ges enregistrements ayant la méme valeur de clé : ils seront “logiquement” voisins et en
ordre Inverse de leur création : le dernler homonyme créé sera le premier lu.

Les programmes P5, P6, P7, P8 et P9 donnent quelques exemples de I'utilisation de
fichlers Indexés : création, écriture, lecture séquentielle ou a la demande. Ici encore, ilne
s'agit que de programmes “pédagogiques". Le programme P8 fall intervenir la commande
XTRACT qui permet d'extraire d'un fichler Indexé ceux des enregistrements dontla valeur
d'une partie de la clé a une certaine valeur. ll existe également une commande BORNE qul
fixe une borne supérieure pour la lecture “séquentielle” : cela veut dire u'll y aura ume
erreur des que I'on arrivera A I'enregistrement en question : cela permet de programmer
facilement la lecture d'une fraction de fichler. Attention : ces deux commandes sont
spécifiques aux fichiers indexés. '

Tous ces exemples travaillent sur des fichiers monoclés. En ce qui concerne les fichiers
multiclés, Il faudrait ‘)réclser, pour certaines commandes, le numéro de la clé (par défaut,
ce serait la premiére

LET "READ-1,2"

permettrait de lire dans le fichier dont l'identificateur temporaire est 1, en utilisant le
tableau d'adresses (ou fichier d'index) relatif 3 la clé numéro 2.

7 CONCLUSION

Nous n'avons pas fait le tour complet des possibilités de MEM/DQOS dans la gestion des
fichiers : celle notice est une introduction ... qui doit éfre poursuivie par la lecture de bro-

chures plus completes dans lesquelles vous trouverez la totalité des commandes de MEM/ (” g
DOS, des explications détaillées sur I'organisation el implantation réelle des fichiers, ...

elc, elc.

8 LES PROGRAMMES PO 4 P9

On trouve pour chaque programine :

a) - un énoncé précis de ce que doit faire ce programme, avec les noms des objets manipu-
lés
)

b) - un algorithine dans lequel on découpe bien chacune des opérations ; on ulilise la struc- (
ture répélitive :

ltérer

. traitement |}

Sorlll sl condition TRAITEMENT 1

. traitement 2

Flnllémr

\\ VRAI

CONDITION

qui correspond a

I organigramme ci-
contre dans lequel
le traitement 1 ou
le traitement 2
peul élre vide ;
dans certains cas, TRAITEMENT 2
on utilise plu-
sieurs tests de
sortie dans la me-
me boucle

c) - un programme

»

d) - des remarques.

- 24 -




PO : CREATION D’UN FICHIER RELATIF

1) Probléme précis :

Créer le fichier relatif FSE dont le numéro de fiche aura comme identificateur NO% et dont
chaque fiche comprendra deux rubriques : un nom NM$ et un solde de compte SC. Avant
d’essayer de créer FSE, il faudra vérifier qu'il n’existe pas déja. Ce test s’ effectue simplement
en essayant d’ouvrir le fichier avant d’essayer de le créer.

2) Algorithme :

. faire le ménage (cf paragraphe 4)
. ouvrir le fichier FSE
. si 'ouverture s'est faite normalement
alors afficher “le fichier FSE existe déja”
sinon décrire la structure d'un enregistrement et le type de FSE
créer le fichier
finsi
. faire le ménage

3) Programme :

10REM : PROGRAMME PO

20LET "#CLEAR,$”

30 LET "# OPEN:1,FICHIER:0-FSE”

40 IF WS = 0 THEN : PRINT "FSE EXISTE DEJA™: GOTO 80
50LET "> @ NO%=NM$,SC”

60 LET "# NEW:1, FICHIER:0-FSE”

70 IF WS = 0 THEN : PRINT "FSE CREE”

80 LET "# CLEAR,$”

90 END

4) Remarques :

1 - Dans ce programme, on a attribué I'identificateur temporaire 1 & FSE et il est créé sur le
lecteur 0.

2 - Une erreur de syntaxe dans la ligne 50 peut se traduire par un message d’erreur en ligne
60, car la description de la ligne 50 n’est utilisée que lors de la création.

3 - La commande LET”# DELETE:FICHIER:0-FSE” supprime FSE.

4 - Comprenez par faire le ménage : nettoyer la mémoire !

_ 95 _



P1: AJOUTER UNE OU PLUSIEURS FICHES AU FICHIER FSE

1) Probléme précis :
Lire au clavier une suite de NM$ et de SC que I'on place dans le fichier FSE.
2) Algorithme :

. faire le ménage
. ouvrir le fichier FSE
. si 'ouverture s’est mal passée alors arréter le programme finsi
. itérer . lire un nom NM$
. sortir si NM$ = "FIN”
. lire le solde correspondant SC
. enregistrer NM$ et SC sur le fichier FSE
. si 'opération s’est mal passée alors arréter le programme finsi
. finitérer
. faire le ménage

3) Programme :

10REM : PROGRAMME P1
20LET "#C,$”

30 LET " # OPEN:1,FICHIER:0-FSE”

40 IF WS THEN : PRINT "FSE N'EXISTE PAS”: GOTO 160
50 REM : ITERER

60 PRINT "DONNEZ UN NOM (FIN=ARRET)”

70 INPUT NM$

80REM : SORTIR SI

90 IF NM$ = "FIN” THEN : GOTO 150

100 PRINT "SOLDE CORRESPONDANT SVP ?”

110 INPUT SC

120 LET "WRITE-1”

130 IF WS THEN : PRINT "ERREUR ECRITURE”: GOTO 160
140 GOTO 50

150 REM : FIN ITERER

160 LET "# C,$”

170 END

4) Remarques :

1 - Le test IF WS est équivalenta IF WS)0 car en BASIC une expression numérique dontla
valeur est nulle est considérée comme fausse.

2 - On n’a jamais eu besoin de préciser le numéro de la fiche NO% : MEM /DOS s’en
occupe tout seul. On peut néanmoins faire afficher ce numéro si on veut.

3 - Le programme peut étre utilisé plusieurs fois de suite : les fiches sont ajoutées au fur et
mesure au bout du fichier.

P2 : LECTURE DE TOUT LE FICHIER RELATIF

1) Probléme précis :

Lire successivement (on dit aussi SEQUENTIELLEMENT) tous les enregistrements du
fichier FSE et les afficher sur I'écran au fur et & mesure.

2) Algorithme :

. faire le ménage
. ouvrir le fichier FSE avec I'identificateur temporaire 4 (pourquoi pas !)
. itérer
. lire Ienregistrement suivant du fichier FSE
. sortir si il n'y a plus d’enregistrement a lire
. afficher les caractéristiques de I'enregistrement lu : numéro, nom, solde
. fin itérer
. faire le ménage

3) Programme :

10 REM : PROGRAMME P2

20 LET"#C,$”

30 LET "# OPEN:4,FICHIER:0-FSE”

40 IF WS THEN: PRINT "FSE N’EXISTE PAS”: GOTO 150
50 REM :ITERER

60 LET "NEXT-4”

70 REM :SORTIR SI

80 IFWS> 0 THEN : PRINT "FIN LECTURE”: GOTO 140
90 PRINT "NUMERO D’ENREGISTREMENT : ";NO%
100 PRINT "NOM: ";NM$

110 PRINT "SOLDE DU COMPTE : ";SC

120 PRINT

130 GOTO 50

140 REM : FIN ITERER

150 LET "# C,$”

160 END

4) Remarques :

R.1:La commande NEXT permet de lire I'enregistrement “suivant” du fichier dont I'iden-
tificateur temporaire est précisé ; et la commande OPEN prépare  la lecture du premier
enregistrement.

R.2: On a vu que WS contient un compte-rendu aprés une opération faisant intervenir la
mini-disquette. Aprés une lecture, si la valeur de ce WS est positive, c’est qu’il n’y avait plus
rien a lire.

R.3 : Notez bien que le seul fait d’ utiliser le fichier FSE permet au programme de connaitre
les noms des identificateurs des “rubriques” d’un enregistrement.

R.4 : Le numéro du premier enregistrement est 1 (cela, on le voit en faisant exécuter le
programme, pas en regardant le texte ci-dessus).

R.5 : On verra plus loin (P.4) qu’on peut facilement lire séquentiellement une partie du
fichier.

P3 LECTURE A LA DEMANDE DE TELLE OU TELLE FICHE

1) Probléme précis :

Lire au clavier un numéro d’enregistrement et afficher le contenu de I'enregistrement
correspondant du fichier FSE. On s’arréte lorsque le numéro demandé est 0.

2) Algorithme :

. faire le ménage

. ouvrir le fichier FSE

. iterer
lire un numéro d’enregistrement : NO%

. sortir si le numéro lu est 0
lire I'enregistrement correspondant dans le fichier
afficher les valeurs lues

. fin itérer

. faire le ménage

3) Programme :

10 REM : PROGRAMME P3

20 LET"#C.$”

30 LET ”# OPEN:4,F.0-FSE”

40 IF WS THEN : PRINT "FSE N’EXISTE PAS”: GOTO 180
50 REM :ITERER

60 PRINT "DONNEZ UN NUMERO (0=FIN)”

70 INPUT NO%

80 REM :SORTIR SI

90 IFNO% = 0 THEN: GOTO 170

100 LET "READ-4”

110 REM: SORTIR SI

120 IF WS THEN : PRINT "ERREUR LECTURE™: GOTO 170
130 PRINT "NOM : ";NM$

140 PRINT "SOLDE DU COMPTE : ”;SC

150 PRINT

160 GOTO 50

170 REM : FIN ITERER

180 LET "#C,$”

190 END

4) Remarques :

R.1: La commande READ permet de lire I'enregistrement dont le numéro est précisé au

préalable. C’est ce que certains appellent 'acces "direct”.

R2:La ligne 120 teste si la lecture s’est bien passée. Avez-vous essayé de lire un enregis-

trement inexistant ?

P4 LECTURE SEQUENTIELLE D’UNE PARTIE D’UN FICHIER RELATIF
1) Probléme précis :

Lire au clavier un numéro d’enregistrement et afficher a Iécran le contenu de cet enregis-
trement du fichier FSE et de tous les enregistrements suivants.

2) Algorithme :

. ouvrir le fichier FSE
. lire le numéro de départ au clavier
. lire I'enregistrement correspondant sur le ﬁchler FSE
. iterer
. sortir si on n’arien lu (fin de fichier)
afficher les valeurs lues
lire I enregistrement suivant
. fin itérer
. fermer le fichier FSE

3) Programme :

10REM : PROGRAMME P4

20LET "#C,$”

30 LET "# OPEN:4,F:0-FSE”

40 IF WS THEN : PRINT "FSE N’EXISTE PAS”: GOTO 180
50 PRINT "NUMERO DE DEPART”

60 INPUT NO%

70 LET "READ-4”

80REM :ITERER

90 REM : SORTIR SI

100 IF WS THEN : PRINT "FIN LECTURE”: GOTO 170
110 PRINT "NUMERO D’ENREGISTREMENT : ”;NO%
120 PRINT "NOM : ”;NM$

130 PRINT "SOLDE DU COMPTE: ";SC

140 PRINT

150 LET "NEXT-4"

160 GOTO 80

170REM : FINITERER

180 LET "# C,$”

190 END

4) Remarques :

R.1: Notez bien la position de I'instruction READ (extérieure a la boucle) qui permet d’ini-
tialiser la lecture.

R.2 : Le programme suivant P5 illustre I'utilisation des fichiers indexés : lisez d’abord le
paragraphe 6 avant de I'étudier : cela vous permettra de faire le point sur les fichiers relatifs
et vous y trouverez également des explications nécessaires a la poursuite de I'étude des
exemples.

P5 CREATION D’UN FICHIER INDEXE MONOCLE

(Avez-vous lu le paragraphe 6 ?)
1) Probléme précis :

Créer un fichier indexé PERSO dont la clé est composée de deux chaines de longueur 10:
NMS$ et PR$, et dont I'enregistrement comprend deux rubriques : un entier appelé IN% et
une chaine appelée AD$.

2) Algorithme :

Sans objet | On utilise le méme principe que dans le programme PO : on essaie d’ouvrir, ce
qui teste I'existence de PERSO !

3) Programme :

10 REM : PROGRAMME P5

20LET "#C,$”

30 LET ”# OPEN:1,FICHIER:0-PERSO”

40 IF WS = 0 THEN : PRINT "PERSO EXISTE DEJA”: GOTO 80
50 LET "»>NM$10,PR$10=IN%,AD$"

60 LET ”# NEW:1,F:PERSO”

70 IF WS = 0 THEN : PRINT "PERSO CREE”

80LET "#C,$”

90 END

4) Remarque :

Un des principes de la méthode ASSIMIL est de s’'imprégner de la langue que I'on
apprend avant d’essayez de la parler ; autrement dit de pratiquer la version avant le
theme. C’est ce que vous avez fait jusqu’ici. Vous pourrez bientét passer a la seconde
phase qui consiste a reprendre ces petits programmes depuis P0, & ne regarder que les
algorithmes et a essayer de les programmer par vous-méme.

P6 ECRITURE DANS UN FICHIER INDEXE MONOCLE
1) Probléme précis :

Enregistrer dans le fichier PERSO une suite de fiches lues au clavier. S’arréter lorsqu’on lit
la valeur “FIN” pour NM$.

2) Algorithme :

. ouvrir le fichier PERSO

. itérer
. lire NM$

. sortir si NM$ = "FIN”
. lire PRS$, IN%, AD$
. enregistrer la fiche

. fin itérer

. fermer le fichier PERSO

3) Programme :

10REM : PROGRAMME P6

20LET"#C,$”

30 LET "# OPEN:1,F:0-PERSO”

40 IF WS> 0 THEN : PRINT "PERSO N’EXISTE PAS”: GOTO 200
50 REM: ITERER

60 PRINT "DONNEZ UN NOM (FIN=ARRET)”
70 INPUT NM$

80 REM :SORTIR SI

90 IF NM$ = "FIN” THEN : GOTO 190

100 PRINT "DONNEZ LE PRENOM”

110 INPUT PR$

120 PRINT "DONNEZ L'INDICE”

130 INPUT IN%

140 PRINT "DONNEZ L’ADRESSE”

150 INPUT AD$

160 LET "WRITE-1”

170 IF WS> 0 THEN : PRINT "ERREUR ECRITURE”: GOTO 200
180 GOTO 50

190 REM: FIN ITERER

200LET " #C,$”

210 END

4) Remarques :

R1 - Sion tolere les homonymes (NM$ et PR$ identiques) il suffiten ligne 160 de remplacer
WRITE par ADD.

R2 - Vous pourrez constater, grace au programme P7, que les fiches sont rangées par ordre
des noms et des prénoms, quel que soit leur ordre d’entrée.

R3 - On ne mentionne plus dans Ialgorithme les tests qui vérifient que tout s’est bien passé
au niveau des transferts sur disquette.

P7 LECTURE SEQUENTIELLE D’UN FICHIER INDEXE MONOCLE
1) Probléme précis :
Lire dans le fichier PERSO tous les enregistrements dont les NM$ sont compris entre deux
bornes qui seront lues au clavier.
2) Algorithme :
. ouvrir le fichier PERSO
. lire au clavier le dernier NM$ et positionner cette borne
. lire au clavier le premier NM$
. se positionner sur I'enregistrement correspondant (il n’y en aura probablement pas : cf
remarque n. 1 ci-dessous)
. itérer

. lire 'enregistrement suivant
. sortir si cela s’est mal passé

. afficher les valeurs lues
. fin itérer
. fermer le fichier PERSO
3) Programme :
10 REM : PROGRAMME P7
20LET "#C,$”
30 LET ”# OPEN:1,F:0-PERSO”
40 IFWS > 0 THEN : PRINT "PERSO N’EXISTE PAS”: GOTO 220
50 PRINT "DONNEZ LE DERNIER NOM”
60 INPUT NM$
70 LET "BORNE-1"
80 PRINT "DONNEZ LE PREMIER NOM”
90 INPUT NM$
100 LET "READ-1”
110 REM :ITERER
120 LET "NEXT-1”
130 REM :SORTIR SI
140 I[F WS> 0 THEN : PRINT "FIN LECTURE”: GOTO 210
150 PRINT "NOM  : ";NM$
160 PRINT "PRENOM : ";PR$
170 PRINT "INDICE : ’;IN%
180 PRINT "ADRESSE: ";AD$
190 PRINT
200 GOTO 110
210 REM: FIN ITERER
220 LET "# C,$”
230 END
4) Remarques :
R.1: On précise la valeur de NM$, mais pas celle de PR$ donc on va essayer de lire un
enregistrement ayant comme clé le NM$ lu et PR$ vide (cf P7 : la clé est composée de deux
variables !) : un tel enregistrement n’existe probablement pas ! Donc la lecture de I'instruc-
tion 100 ne va pas aboutir, ce que I'on pourrait vérifier par un

105 PRINT WS

Voila pourquoi I'algorithme est différent de celui du P4.
R.2: La méme analyse est a faire pour I'arrét.
R.3: On aurait, bien sr, pu lire les PR$ au clavier.

P8 EXTRACTION DES ENREGISTREMENTS DONT UNE PARTIE DE LA CLE
SATISFAIT A UNE CONDITION DONNEE

1) Problémé précis :

Lire dans le fichier PERSO tous les enregistrements dont la valeur de PR$ est égale & une
valeur lue au clavier.

2) Algorithme :

. ouvrir le fichier PERSO
. lire la valeur de PR$
. prévenir que I'on veut extraire & partir de la deuxiéme rubrique de la clé
. itérer
. lire 'enregistrement suivant (ayant la borne valeur de PR$)
. sortir si on n'a pas lu... car il n’y avait plus d’enregistrement satisfaisant
. afficher les valeurs lues
. fin itérer
. fermer le fichier PERSO

3) Programme :

10REM : PROGRAMME P8
20LET "#C,$”

30 LET "# OPEN:1,F:0-PERSO”
40 IF WS> 0 THEN : PRINT "PERSO N’EXISTE PAS”: GOTO 200
50 PRINT "DONNEZ LE PRENOM”
60 INPUT PR$

70 LET "XTRACT-1,2”

80 LET "READ-1”

90 REM :ITERER

100 LET "NEXT-1"

110 REM :SORTIRSI

120 IF WS> 0 THEN : PRINT "FIN LECTURE”: GOTO 190
130 PRINT "NOM  : ";NM$

140 PRINT "PRENOM : ”’;PR$

150 PRINT "INDICE : ";IN%

160 PRINT "ADRESSE: ”;AD$

170 PRINT

180 GOTO 90

190 REM : FINITERER

200LET "#C,$”

210END

4) Remarques :

R.1 : On aurait pu extraire de la méme fagon tous les enregistrements ayant un certain
NMS$ : il aurait suffit, en ligne 70, de le préciser : LET”XTRACT-1,1” sans oublier de modi-
fier les lignes 50 et 60 en conséquence !!! Mais en fait cette “extraction” pourrait alors étre
faite plus rapidement grace a la commande BORNE puisque tous les enregistrements
ayant le méme NM$ sont consécutifs.

R.2 : Les mémes remarques concernant le non-aboutissement de la lecture de la ligne 80
sont a faire (cf P7).

- 33 -




P9 CREATION D’UN FICHIER AYANT UNE RUBRIQUE DATE
1) Probléme précis :

Créer un fichier indexe PERSODATE dont la clé est une chatne SS$ de 13 caracteres et ( /
dont chaque enregistrement est formé de deux rubriques : une chaine NM$ et une date
DAS$*. Enregistrer ensuite des fiches (on s’arréte lorsque SS$ lu vaut 000)

2) Algorithme :

. décrire la structure de PERSODATE
. créer ce fichier
. itérer lire SS$
. sortir si la valeur lue est 000"
. lire le NM$ et le DA$ correspondants
. enregistrer la fiche
. fin itérer
. fermer le fichier PERSODATE (
3) Programme :

10REM : PROGRAMME P9
20LET "#C,$”

30 LET ” # OPEN:1,FICHIER:0-PERSODATE”
40 IFWS = 0 THEN : GOTO 70

50 LET ”»SS$13=NM$,DA$*”

60 LET ”# NEW:1,F:0-PERSODATE”

70 REM :ITERER

80 PRINT "NUMERO SS (000=ARRET)

90 INPUT SS$

100 REM : SORTIR SI

110 IF SS$ = “000“ THEN : GOTO 190

120 PRINT "DONNEZ LE NOM*

130 INPUT NM$

140 PRINT "DONNEZ LA DATE*

150 INPUT DA$

160 LET "WRITE-1*

170 IE WS> 0 THEN : PRINT "ERREUR ECRITURE": GOTO 190 (
180 GOTO 70

190 REM : FIN ITERER

200 LET ”#C,$"

210 END

4) Remarques :

R.1: Notez bien que la variable de type DATE a comme identificateur DA$ dans I'instruc-
tion 150, mais DA$* dans la commande 50. (Cf annexe II).
R.2: Les lettres D et A sont évidemment des lettres quelconques !!!

R.3: Essayez quelques dates ... pour voir ce que comprend la machine vous avez peut-étre
intérét a rajouter deux instructions.

172 LET"READ-1¢ | (
173 PRINT"DATE ENREGISTREE: ”:DA$

Chapitre III

UTILISATION DES MASQUES

2 COMMENT CREER UN MASQUE

2 -2 Comment décrire un masque ?

3 UTILISATION D’UN MASQUE EN SAISIE

1 DEFINITION ET UTILISATION DES MASQUES

Dans le vocabulaire courant, un masque est une “chose” que des personnes différentes
peuvent se mettre devant le visage : elles auront toutes les mémes traits, mais a I'endroit
des trous (yeux, bouche) ce que I'on verra pourra étre différent selon la personne qui porte
le masque a cet instant.

En informatique, un masque est un “objet” qui est Ccomposé :

- d’un texte fixe (c’est le masque)

- de “fenétres” (ce sont les trous ci-dessus)

- d'identificateurs de variables (I'équivalent des “yeux”, “bouche” ci-dessus) : il y a autant
de variables que de fenétres.

Avant d’utiliser un masque, il faudra le créer (et le mémoriser sur disque le plus souvent).
Ensuite, il pourra servir en entrée ou en sortie.

En entrée (pour la saisie des données), le texte fixe du masque est projeté sur I'écran et
I'utilisateur entre les données dans les fenétres (en tapant les valeurs au clavier) puis
I'ensemble des variables dont on vient ainsi de précisef les valeurs est utilisé : le plus sou-
vent ces variables constitueront un enregistrement et cet enregistrement sera envoyé dans
un fichier.

En sortie, I'utilisation d’'un masque permet de faciliter I'affichage sur écran ou les impres-
sions grace, en particulier, & la notion de FORMAT DE SORTIE (cela n’a rien a voir avec le
formattage d’une disquette ; pour ceux qui n’ont pas cette notion, patience jusqu’au para-
graphe 5).

L'utilisation des masques sera illustrée a I'aide du fichier PERSO, créé a I'aide du pro-
gramme P5, qui rappelons-le, a une clé composée de deux variables NM$ et PR$ et a deux
rubriques IN% et AD$.

2 COMMENT CREER UN MASQUE

2 - 1 Principe de I'opération

On vient de voir qu'un masque est un objet. Il pourra donc étre mémorisé sur le disque
sous un certain nom (identificateur) qui figurera au catalogue sous la rubrique M. Cet objet
est un ensemble d’informations : le texte fixe, les emplacements des fenétres, les identifica-
teurs des variables correspondant a chaque fenétre. La création d’'un masque (MSE par
exemple) passe par trois étapes :
- on exécute la commande
LET“# NEW:1,MASQUE:MSE”

qui a pour effet de passer la main a I'utilisateur (le masque créé s’appellera MSE)
- on nettoie I'écran en tapant SHIFT-CTRL-@
- on décrit alors le masque sur I’écran grace aux commandes qui seront détaillées ci-dessous,
- on rend la main & MEM/DOS (en tapant ESC) ce qui a pour effet de sauvegarder le

masque MSE. '

- 36 —




2 - 2 Comment décrire un masque

Le masque MSE est reproduit ci-dessous :

R R RS R R R R R R R S R R R R e R R B

moan "NMs
AN “PRS$” >

D I
"aps "

TAPER RETURN POUR
PASSER AU SUTUANT
rrtx

e RS e R aF R e R RO OS] TXEXEXEXLLLEXRL

%
3
X
3
3
3
3
g
%
3
3
bs
z
3
b
p
X
b3
b3
3
3
b3
X
X

Il comprend 4 “fenétres” dont les positions sont repérées par les caractéeres < et > et 4
identificateurs de variables qui apparaissent dans ces fenétres. Le reste, c’est le “texte” (y

compris les caractéres * qui forment un cadre ... et qui ne sont évidemment 1a que pour
faire plus joli !).

REMARQUES :

R. 1: La position et la largeur des fenétres sont arbitraires, mais la largeur doit étre compati-
ble avec le nombre de caractéres de la variable qui y figure (et qui est malencontreusement
appelée le plus souvent : “longueur”) : ici, le NM$ ayant été déclaré a la création du fichier
PERSO avec une “longueur” 10, la “largeur” de la fenétre correspondante devrait étre
logiquement de 10. Précisons & ce sujet que < et> comptent dans le décompte ... etque, en
cas de manque de place I'identificateur peut étre écrit a I'extérieur de la fenétre. Silafenétre
n’a qu’un caractére de large, il faut la positionner grace au seul caractere >.

R. 2: Les identificateurs des variables apparaissent entre guillemets. Ne les oubliez surtout
pas ! Ou alors MEM/DOS refusera de mémoriser votre masque lorsque vous taperez ESC
(il vous rappelera a I'ordre par un coup de sonnette).

R.3: Le caractére " ne doit absolument pas apparaitre dans le texte du masque : il serait
tres mal digéré par MEM/DOS ; en effet, on a vu ci-dessus qu’un identificateur de variables
entoura de ses ” pourrait, en cas de manquede place, apparaitre a I’ extérieurde sa fenétre;
I'apparition d’un ” oit qu'il soit dans I'écran, est donc analysé par MEM/DOS comme le
debut d'un identificateur de variable. Pour la méme raison, les caractéres € et> ne doivent
pas intervenir dans le texte.

R.4 : En fait, les identificateurs des variables peuvent apparaitre n’importe ot ! Lorsque
MEM/DOS analyse un masque, il le balaye ligne par ligne de haut en bas et constitue, au
fur et & mesure qu'il les rencontre, une suite de fenétres et une suite d'identificateurs de
variables ; il met en correspondance ces deux suites : la premiére fenétre rencontrée avec le
premier identificateur, et ainsi de suite.

Maintenant que I'on sait ce que I'on doit faire, voyons comment le faire : essentiellement
comment décrire un tel masque dans un écran vide aprés avoir exécuté la commande

LET ”#NEW:1,MASQUE:0-MSE”

Le curseur se déplace horizontalement grace aux fléches du clavier et verticalement par
CTRL-R (haut) et CTRL-Z ou RETURN (bas). Pour répéter un caractére sur une méme
ligne, il suffit sur APPLE Il d'utiliser la touche REP et sur APPLE Ile. de maintenir la
pression, comme d’habitude. Mais en plus, sur les deux versions, on peut répéter un
caractére verticalement vers le bas en plagant le curseur sur ce caractére et en tapant
CTRL-V. Les caractéres peuvent étre écrits normalement, ou inversés ou clignotants grace
a I'utilisation de CTRL-N, CTRL-B (pour fond Blanc) ou CTRL-F (pour Flash). Enfin, on
peut effacer tout I'écran par SHIFT-CTRL- @

Vous pourrez vous entrainer & créer le masque MSE (ou n’importe quel autre). N’oubliez
pas de faire ESC quand la description est terminée ... et si vous entendez alors une
sonnette ... c’est peut-étre que vous avez oublié de mettre les variables entre .

3 UTILISATION D’UN MASQUE EN SAISIE

Supposons que le masque MSE soit créé et que I'on veuille I'utiliser pour saisir des
données et les enregistrer dans le fichier PERSO. Il faut écrire un programme dans lequel
on commence par “ouvrir” le masque (en lui attribuant un identificateur temporaire : B par
exemple) par LET” # OPEN:B,MASQUE:0-MSE” puis on le prqjette sur I'écran grace a la
commande

LET"CHARGE-B”
Ensuite on peut saisir plusieurs enregistrements : la commande
LET”INPUT-B”

saisit I'ensemble des variables du masque : lors de I'exécution, I'utilisateur doit remplir les
fenétres du masque et taper ESC lorsqu'’il les aura correctement remplies ; la touche
RETURN permet simplement de passer a la fenétre suivante ... et la derniére fenétre est
“suivie” par la premiére, ce qui permet de faire des corrections et ce qui explique que la
touche RETURN ne peut pas servir i la saisie de I'ensemble des variables : c’est la touche
ESC qui remplit cette fonction. Le programme P10 illustre ces notions. La commande

LET”PRINT-B”

permet a tout instant de visualiser le texte et les valeurs qu’ont actuellement les variables.
C’est une commande particuliérement utile lorsqu’on veut faire modifier par I'utilisateur
une ou plusieurs de ces valeurs : on exécute succesivement une commande PRINT et une
commande INPUT. La commande )

LET"OUTPUT-B”

affiche sur I'écran les valeurs des variables a I'emplacement de leurs fenétres respectives,
mais sans modifier le reste de I'écran et donc sans afficher le texte du masque. Enfin, la
commande

LET"VISUALISE-B”

affichera simplement le masque complet : texte, fenétres et identificateurs des variables.

4 UTILISATION D’UN MASQUE EN SORTIE ECRAN OU PAPIER

Le masque MIMPRIM est utilisé dans le programme P12 de sortie sur imprimante, alors
que le programme P11 de sortie sur écran utilise le masque MSORTIE. Ces masqges ont été
créés pour vous sur la disquette MEM/DOS afin que vous puissiez essayer directe-
ment les programmes P11 et P12.

5 LES FORMATS DE SORTIE

Le BASIC de I'APPLE ne permet pas d'obtenir facilement des affichages “élégants” de
tableaux de noms ou de valeurs numériques : il est difficile d’arrondir des valeurs & un
nombre donné de chiffres derriére la virgule par exemple.

On appelle format de sortie I'ensemble des renseignements permettant de décrire comment
on veut présenter une information : nombre total de cases (ou positions), position de la
virgule (donc nombre de chiffres deriére la virgule), calage a gauche ou a droite dans
I'emplacement réservé. Cette possibilité de préciser un format de sortie existe sur certains
BASIC ; elle est alors souvent appelée PRINT USING.

Les masques de MEMDOS permettent, tant pour I'affichage sur I'écran que pour I'impres-
sion, un “formattage” des valeurs numériques : il y a dans MEM/DOS trois formats possi-
bles et le choix s’effectue lorsqu’on précise I'identificateur a afficher dans la fenétre de
'écran ; par “défaut”, c’est le format classique du BASIC (cadrage & gauche, quantités
écrites comme d’habitude) ; si la variable est suivie des caractéres a * (“FL a *” par
exemple) la valeur sera cadrée a droite et les quantités écrites comme d’habitude ; enfin, si
la variable est suivie de * seulement (“FL*” par exemple) les valeurs sont cadrés a droite
avec un nombre constant de chiffres derriére la virgule : c’est ce que I'on appelle le format
“gestion” ; par défaut, il y aura 2 chiffres derriére la virgule, mais ce nombre peut étre
modifié par la commande

LET” @ -3”

par exemple, si on veut trois chiffres au lieu de deux.

6 CONCLUSION

Nous ne pouvons que reprendre les termes de la conclusion du chapitre Il : ce que vous
avez vu dans cette notice n’est qu’une introduction a I' utilisation des masques. En parti-
culier en ce qui concerne I utilisation de tableaux dans les masques (ou les fichiers), les
caches dans les masques, les sorties sélectives, I'impression ligne a ligne... etc ... Grace
a ces outils, vous améliorerez a la fois la quallte de vos programmes (hsnbnhte rapldlte) et
votre producnwte

7 LES PROGRAMMES P10 A P12

On trouvera dans les pages suivantes les programmes P10 a P12

P10 UTILISATION D’UN MASQUE EN SAISIE DE DONNEES

(ATTENTION: si vous n’avez pas créé le masque MSE, créez le, ou exécutez le programme
PSECOURS qui crée MSE a partir du masque MSECOURS qui existe d’origine).

1) Probléme précis :

Grace au masque MSE, saisir plusieurs enregistrements destinés au fichier PERSO et les y
charger. S’arréter lorsqu’on lit FIN pour NM$.




2) Algorithme :

Ouvrir le masque MSE et le fichier PERSO.
Afficher le texte du masque MSE.
. itérer
. lire au clavier les valeurs du masque MSE
. sortir si NM$ = “FIN” ou si Iutilisateur abandonne
. écrire |'enregistrement dans le fichier PERSO
. fin itérer
. Effacer I’écran
. Fermer tous les objets.

3) Programme :

10REM : PROGRAMME P10

20LET " #C,$”

30 LET ”# OPEN:M,MASQUE:0-MSE”

40 IF WS THEN : PRINT "ERREUR OUVERTURE MSE™: GOTO 160
50 LET ”# OPEN:F,FICHIER:0-PERSO”

60 IF WS THEN : PRINT "ERREUR OUVERTURE PERSO”: GOTO 160
70 LET "CHARGE-M”

80REM :ITERER

90 LET "INPUT-M”

100 REM :SORTIR SI

110 IF NM$ = "FIN” OR WS THEN: GOTO 140

120 LET "WRITE-F”

130 GOTO 80

140 REM:FIN ITERER

150 HOME

160 LET “ #C,$”

170 END

4)Remarques :

R.1: Les identificateurs temporaires des objets doivent évidemment étre différents, méme
sl s’agit d’objets de type différent. Ici, on a choisi M pour le masque et F pour le fichier,
mais ce n’est pas obligatoire.

R.2 : L’instruction CHARGE étant placée a I'extérieur de la boucle itérer, quand on tape
ESC pour enregistrer le masque puis la fiche et que I'on revient a I'instruction 90, les
valeurs des variables restent affichées, ce qui ac¢élere la saisie s'il y a des variables a
conserver pour |'enregistrement suivant.

R.3 : Il est dangereux de mettre I'instruction CHARGE dans la boucle itérer : ce n’est pas
parce que les valeurs des variables ne seraient plus affichées qu’elles seraient remises a
zéro ou a blanc : I'utilisateur pourrait croire qu’il saisit une fiche en laissant des zones vides
alors que ces zones auraient la valeur de la fiche précédente ! Par contre, on peut mettre un
LET”PRINT-M” dans la boucle.

P11 UTILISATION D’UN MASQUE POUR SORTIE ECRAN
1) Probléme précis : {

Grace au masque MSORTIE, afficher séquentiellement tous les enregistrements du fichier
PERSO. Aprés I'affichage d’un enregistrement, attendre que !'utilisateur tape RETURN
pour afficher le suivant.

2) Algorithme :

Ouvrir le masque MSORTIE et le fichier PERSO.

. Afficher le masque vide.

. iterer
. lire I'enregistrement suivant de PERSO

. sortir si on n’arien lu
. afficher les valeurs sur le masque (sur I'écran) (
. attendre que I' utilisateur tape un caractére

. fin itérer

. Effacer I'écran

. Fermer les objets

3) Programme :

10 REM : PROGRAMME P11

20LET” C¢$”

30 LET "# OPEN:1,M:0-MSORTIE”

40 LET " # OPEN:2 F:0-PERSO”

50 REM: ITERER

60 LET "CHARGE-1"

70 LET "NEXT-2”

80 REM :SORTIR SI

90 IF WS> 0 THEN : GOTO 130

100 LET "OUTPUT-1”

110 GET C$ r
120 GOTO 50 (
130 REM: FIN ITERER

140 HOME

150 LET "#C,$”

160 END

4) Remarques :

R.1 : On lit un caractere (GET) plutét qu’autre chose (INPUT) afin de ne pas décaler le
masque sur I’écran d’une ligne vers le haut. On aurait pu tout aussi bien utiliser un masque
uniquement pour cette saisie. C’est ce qui sera fait dans I'exemple du chapitre V.

R.2 : 1l serait prudent de rajouter des instructions vérifiant que les ouvertures se sont bien (
effectuées.

P12 UTILISATION D’UN MASQUE POUR IMPRESSION

1) Probléme précis :

Grace au masque MIMPRIM afficher et imprimer séquentiellement les enregistrements du
fichier PERSO. Aprés chaque impression, on attend la lecture d'un caractére et on s’arréte
d’imprimer si ce caractére est A.

2) Algorithme :

. Ouvrir le masque MIMPRIM et le fichier PERSO
. iterer
. lire 'enregistrement suivant de PERSO
. sortir si on n'arien lu
. afficher le masque sur I'écran avec ses valeurs
. recopier I'écran sur I'imprimante
. lire un caractére au clavier
. sortir si le caractére lu est “A”
. fin itérer
. Effacer I'écran
. Fermer les objets

3) Programme :

10REM : PROGRAMME P12
20LET"#C$”

30 LET " # OPEN:4,M:0-MIMPRIM”

40 LET ” # OPEN:2,F:0-PERSO”

50 REM : ITERER

60 LET "NEXT-2”

70 REM : SORTIR Sl

80 IF WS THEN: GOTO 190

90 LET "CHARGE-1"

100 LET "OUTPUT-1"

110 PR # 1: POKE 33,1: POKE 35,1: VTAB 1: HTAB 1: LET "2-*""
TEXT:PR#0

120 HOME

130 PRINT "TAPEZ A POUR ARRETER”
140 PRINT "OU RETURN POUR CONTINUER”
150 GET LU$

160 REM :SORTIR SI

170 IF LU$ = "A” THEN: GOTO 190
180 GOTO 50

190 REM :FIN ITERER

200 HOME

210LET "#C$”

220 END

* attention, voir les remarques page 44.

4) Remarques :

‘R.1: L'instruction 110 nécessiterait un long développement pour étre clairement expliquée.
Utilisez-la comme une recette de cuisine : elle recopie I'écran sur I'imprimante connectée ( i
sur le slot 1.

R.2 : Le masque vide a été placé dans la boucle a cause des messages 130 et 140 qui
décalent I'écran.

R.3: ll existe une commande LET"PRINT-1" qui a le méme effet qu'un CHARGE-1 suivi
d’'un OUTPUT-1. Attention, ce n’est pas parce que cette commande s’appelle PRINT
qu’elle imprime: non, cette commande affiche surI’écran. C’est pour éviter cette confusion
qu’on ne I'a pas utilisée dans ce programme, mais on peut le faire.




CHAPITRE IV :

LES SOUS-PROGRAMMES

1 INTRODUCTION

2 DEFINITION D'UN SOUS-PROGRAMME ET APPEL
3 EXEMPLES SIMPLES

4 RECURSIVITE ET IMBRICATION

5 LES SOUS-PROGRAMMES SP0 a SP5

1 INTRODUCTION

Une des clés de I'efficacité en informatique, comme ailleurs, réside dans la réduction des
difficultés par le fractionnement gréce a la technique de la programmation modulaire. Les
langages les moins évolués contiennent la notion de sous-programme : en BASIC la
fameuse paire GOSUB, RETURN permet d’éviter la répétition d’une suite d’instructions
utilisée en plusieurs endroits. MEM/DOS permet d’aller beaucoup plus loin dans la modu-
larisation car il ajoute au BASIC des notions réservées jusqu’alors aux langages les plus
modernes (PASCAL, ADA) : parametres formels et effectifs, variables locales et globales,
récursivité !

Avant de décrire I' utilisation des sous-programmes avec MEM/DOS, nous allons rappeller
brievement le principe méme de I'utilisation des sous-programmes et quelques définitions.

Un programme est une suite d’instructions que vous pouvez faire exécuter une ou
plusieurs fois en changeant éventuellement les conditions d’exécution grace a la lecture de
certaines valeurs. Un sous-programme est, de la méme fagon, une suite d’instructions qui
peut étre exécutée une ou plusieurs fois en changeant éventuellement certaines conditions
d’exécution ; mais la différence essentielle est qu’ici, I'exécution est déclenchée automati-
quement par une instruction spéciale CALL placée au bon endroit dans le programme qui
utilise le SP et que I'on appelle le programme appelant ou programme principal. Comme
pour un programme, I'utilisation d’un sous-programme se décompose donc en deux
phases : la description du traitement qui est faite une fois pour toutes, et I'exécution de ce
traitement. Nous allons voir maintenant que les échanges d’informations entre le SP et le
programme principal peuvent se faire de plusieurs facons différentes dont il est important
de bien comprendre le pourquoi et le comment.

Variables globales et variables locales.

Dans un programme, la valeur des variables peut étre fixée par affectation du résultat du
calcul d’une expression ou par une lecture (au clavier, dans un fichier ou en mémoire
centrale par un PEEK) ; dans un sous-programme, la valeur d’'une variable peut aussi
provenir du programme appelant : si une variable EX par exemple est utilisée dans un
sous-programme et que la valeur 12 lui a été affectée lors de I'exécution du programme
principal avant 'appel du SP, cette valeur 12 est utilisée dans le sous-programme ot elle
peut d’ailleurs étre modifiée ; aprés I'exécution du SP la valeur acquise reste affectée a la
variable EX et peut étre utilisée dans la suite du programme principal. Une telle variable est
appelée globale.

On sent immédiatement le besoin d’une autre espéce de variable : en effet, si par exemple
I'appel d'un SP se fait depuis I'intérieur d’'une boucle gérée par un indice K variantde 1aN
et si dans le SP la valeur de K est modifiée, vous imaginez facilement ce qui va se passer :
avant le premier appel, K vaut 1 mais aprés exécution du SP, K ne vaut plus 1, donc la
boucle du programme principal aura un fonctionnement pour le moins inattendu !!! La
notion de variable locale répond a cette préoccupation ; si une variable K est déclarée
locale lors de la description du SP (et on verra plus loin comment), alors pour la durée de
I'exécution de ce SP, le systeme réserve un emplacement en mémoire pour cette variable,
s'en sert et libére la place en fin d’exécution du SP. S'il existe dans le programme appelant
une variable K, il y a une distinction totale entre les deux variables K : aucune valeur ne
passe de 'une a l'autre : ce sont deux objets distincts.

MEM/DOS permet cette distinction entre variable locale et variable globale ; nous verrons
comment préciser les choses au paragraphe 2.

Parameétres formels et paramétres effectifs.

Si on ne pouvait utiliser que des variables locales ou globales pour la description du
traitement a effectuer par le SP, on serait limité : une méme variable devrait avoir le méme
nom lors de la description du SP et lors de son utilisation ; or, pour une certaine variable, on
peut &tre amené a décrire le traitement en lui donnant un nom, X eta utiliser ce SP succes-
sivement pour une variable qui s’appelle Y dans le programme appelant puis une autre qui
s’appelle Z, c’est ce que vous faites couramment lorsque vous utilisez en Basic une fonction
comme SIN(Y) ou SIN(Z). D’ou la notion de “paramétre formel” X et de “paramétre
effectif” Y ou Z. Un parametre formel est un objet servant lors de la description de la
“forme” du SP et un parameétre effectif est celui sur lequel le traitement est “ effectivement
effectué”. Le mot argument est parfois utilisé comme synonyme de paramétre. Ici encore,
MEM / DOS vous permet d’utiliser cette notion de paramétre formel et de paramétre
effectif.

2 DEFINITION D’UN SOUS-PROGRAMME ET APPEL

La définition d'un sous programme en MEM/DOS étend le réle de I'instruction DEF FN :
rappelons d’abord qu’en Applesoft I'exécution de la séquence

10DEFFNA(W) =2*W + 3
20 PRINT FN A(4)

a pour résultat d’afficher la valeur 11. Les deux restrictions principales de cette instruction

en Applesoft sont que la fonction doit étre définie en une seule ligne et qu'il ne peut y avoir
qu’un seul parametre et qu’il doit étre de type numérique.

Sous MEM/DOS. la définition d’un sous-programme se fait de la facon suivante :

DEF FN “nom du sous-programme” (liste d’identificateurs de variables)

... suite d’instructions

END FN

Le nom du sous-programme estlimité & 20 caractéres. La liste d’identificateurs de variables
précise la suite des parameétres formels suivie de la liste des variables locales.

[’exécution d’un sous-programme est déclenchée par I'instruction :
CALL FN “nom du sous-programme” (liste des paramétres effectifs).

I faut remarquer qu'il ne faut pas répéter la liste des variables locales a la suite de la liste des
parametres effectifs. C’est inutile, et de plus c’est par cette différence que MEM / DOS
sépare les paramétres et les variables locales. Précisons par ailleurs que la transmission des
paramétres se fait par adresse : MEM/DOS attribue au paramétre effectif le méme empla-
cement mémoire (la méme adresse) que le paramétre formel : donc toute modification de
la valeur du paramétre formel affectera, apres la fin du sous-programme, le paramétre
effectif. Cela entraine également que les parametres effectifs ne peuvent étre que des
variables.

Lors de la définition, nous vous conseillons, mais ce n’est pas obligatoire, d’utiliser un
séparateur différent pour isoler la liste des paramétres de la liste des variables locales et un
autre séparateur pour distinguer les parametres d’entrée des parametres de sortie, méme si
ces séparations n’ont aucun effet réel. On propose

sl,s2,...=el,e2,../11,12 ...

les identificateurs s1, s2 représentant les parametres dont la valeur résulte d’un traitement
effectué dans le sous-programme, les identificateurs el, e2, ... ceux dont la valeur est
connue a I'appel du sous-programme et les identificateurs 11, 12, ... désignant les variables
locales. Un sous-programme peut n’avoir ni paramétres, ni variables locales;; il se comporte
alors comme une SUBROUTINE ; mais un sous-programme ne peut avoir de variables
locales s’il n’a pas de parametres (on peut tourner cette restriction en déclarant un
parametre “bidon”).

3 EXEMPLES SIMPLES

L’exemple SPO permet d'illustrer les distinctions de base entre paramétres formel et
effectif, d’entrée et de sortie, les variables locales et globales.

L’exemple SP1 montre que MEM /DOS ne fait pas de distinction entre les parameétres
d’entrée et de sortie : il les transmet par adresse et donc, si un parametre dit d’entrée est
modifié dans le sous-programme, sa valeur modifiée est transmise au programme principal.
Donc MEM / DOS ne fait pas la difference qui existe en PASCAL entre parametres
“valeur” et parametres “variables” : tous les paramétres MEM/DOS sont “variables” au
sens PASCAL.

L’exemple SP2 montre que des tableaux & une dimension peuvent figurer comme
paramétres ; leurs identificateurs sont alors suivis d’un point-virgule et leurs tailles doit étre
précisées par une instruction DIM dans le programme appelant. Par contre un tableau ne
peut pas figurer dans la liste des variables locales.

4 RECURSIVITE ET IMBRICATION

Il n’existe probablement pas de texte sur la récursivité qui ne cite I'exemple de la factorielle.
Vous trouverez la version MEM/DOS dans I'exemple SP3. En ce qui concerne I'imbrication,
précisons qu’un sous-programme peut en appeler un autre mais qu’un sous-programme
ne peut pas étre défini dans la définition d’un autre sous-programme ; autrement dit, la
notion de localité n’existe pas pour les sous-programmes. Enfin, une des grosses sources
d’erreurs difficiles a détecter dans les programmes comportant plusieurs modules provient
du pari que font les programmeurs en ce qui concerne le mode de recherche des variables
globales lors d’appels imbriqués : 'exemple SP4 montre que cette recherche est faite
dynamiquement, c’est-a-dire selon la séquence d’appel ; contrairement au PASCAL o1
cette recherche est faite statiquement : selon la séquence de définition.

L’exemple SP5 montre la derniére facilité que vous procure MEM/DOS (qui se montre la
supérieur au PASCAPPLE) : un sous-programme peut-étre paramétre d'un autre sous-
programme. On verra au chapitre V que des fichiers peuvent également étre parameétres ce
qui est strictement interdit en PASCAL.

LES SOUS-PROGRAMMES SPO a SP5

SPO : PRESENTATION DES DIFFERENTES NOTIONS

1) Probléme précis :

Ecrire un sous-programme FACT qui calcule de maniére itérative la factorielle de X (c’estle
produit des X premiers entiers). S’en servir dans un programme principal pour calculer la
factorielle de différents entiers. Le symbole de la fonction factorielle est le point d’exclama-
tion et par convention 0! = 1.

2) Liste du programme :

10REM : ITERER

20 INPUT “DONNEZ UN ENTIER POSITIF”;N%
30REM : SORTIR SI

40 IF N% < = 0 THEN : PRINT "AU REVOIR”: GOTO 90
50 PRINT "OK, VOICI SA FACTORIELLE...”

60 CALL FN "FACT”(RES=N%)

70 PRINT N%,;”!=";RES: PRINT

80 GOTO 10

90 REM :FIN ITERER

100 END

110 DEF FN "FACT”(Y = X%/K)

120 IF X% < 0 THEN : PRINT "ERREUR”: GOTO 170
130Y=1

140FORK = 1 TO X%

150Y=Y*K

160 NEXT K

170END FN

3) Remarques :

R1 - RES, N% sont les parametres effectifs correspondant aux parametres formels Y et
X%. La variable K est une variable locale.

R2 - Les sous-programmes doivent étre placés hors du programme principal.

R3 - La valeur d’une factorielle est toujours un entier, mais on le calcule comme un réel car
c’est une fonction qui crofit trés vite :

DONNEZ UN ENTIER POSITIF 12

OK, VOICI SA FACTORIELLE...

12!=479001600

DONNEZ UN ENTIER POSITIF 13

OK, VOICI SA FACTORIELLE...

13!=6.2270208E+09

SPI : ETUDE DE LA TRANSMISSION DES PARAMETRES

1) Probléme précis :

Ecrire un sous-programme qui utilise et modifie les deux paramétres formels S et E et
montrer en I'utilisant que les valeurs de ces paramétres sont bien transmises a I'entrée et a
la sortie du sous-programme. Vérifier également qu’une variable locale perd sa valeur a la
sortie du sous-programme.

2) Liste du programme :

10 LOCALE = 55:GLOBALE = 2

20 CALCULEE = 10

30 ENTREE = 100

40 CALL FN "SE”(CALCULEE = ENTREE)

50 PRINT "SORTIE DU SOUS-PROGRAMME”

60 PRINT "ENTREE=";ENTREE;” CALCULEE=";CALCULEE
70 PRINT "LOCALE=";LOCALE;” GLOBALE=";GLOBALE
80 END

90 DEF FN "SE”(S = E/LOCALE)

100 LOCALE = 1

110 PRINT "S=";§

120S =S + LOCALE + E

130 PRINT "S=",S

140 FORK = 1 TO GLOBALE

150 PRINT "E=";E

160E = E + LOCALE + S

170 NEXTK

180END FN

3) Remarques :

R1 - La maniére efficace d’étudier cet exemple consiste a le faire exécuter & la main, ce qui
revient a remplir les cases ci-dessous :

S =
S =
E =
E=
SORTIE DU SOUS-PROGRAMME
ENTREE= CALCULEE=
LOCALE= GLOBALE=

R2 - Faites exécuter et comparez. En cas de différence, retour au paragraphe IV.1.

SP2 : UTILISATION DES TABLEAUX COMME PARAMETRES

1) Probléme précis :

Ecrire un sous-programme SOMTAB qui calcule la somme S des N éléments d’un tableau
T. L'utiliser pour calculer la somme TT des NB éléments d’un tableau U dans lequel on a
placé les NB premiers entiers.

2) Liste du programme :

10 INPUT NB

20 DIM U(NB)

30FORK =1TONB

40 U(K) =K

50 NEXT K

60 CALL FN "SOMTAB”(TT = U;,NB)
70PRINTTT

80 END

1000 DEF FN "SOMTAB”(S = T;,N/K)
10105 =0

1020 FORK=1TON

1030S =S + T(K)

1040 NEXT K

1050 END FN

3) Remarques :

R1 - Le tableau doit étre dimensionné dans le programme principal.

R2 - Les identificateurs de tableaux U et T sont suivis d’'un ;.

R3 - 1y a deux variables K : une globale et une locale : il n’y a pas de collision.

R4 - Dans l'instruction 60 n'appellez surtout pas TOTAL la somme des éléments du
tableau U. (TO est un mot réservé du Basic)

SP3 : CALCUL RECURSIF DE N!

1) Probléme précis :

La définition de N! donnée dans I'exemple SPO montre que cette fonction peut étre définie
récursivement :

SIN=0
ALORSN! =1
SINON N!'= N*(N-1)!
FINSI

Le probleme est donc d’écrire un sous-programme basé sur cette définition récursive.

2) Liste du programme :

10REM : ITERER

20 INPUT "DONNEZ UN ENTIER POSITIF : ”;N%

30REM :SORTIR SI

40 IF N% < 0 THEN : PRINT "TANT PIS, J’AVAIS DIT POSITIF !!!”: GOTO 100
50 PRINT "MERCI, JJEN CALCULE LA FACTORIELLE.....": PRINT
60 CALL FN "FACT”(RES=N%)

70 PRINT N%;”!=";RES: PRINT

80 GOTO 10

90 REM : FINITERER

100 END

1000 DEF FN "FACT”(Y = X%/K%,U)

1010 IF X% < 0 THEN : PRINT "ERREUR”: GOTO 1100

1020IF X% = 0 THEN: Y = 1: GOTO 1100

1030 K% = X% - 1

1040 CALL FN "FACT”(U=K%)

1050Y = U * X%

1100 END FN

3) Remarques :

R1 - Les résultats sont identiques a ceux de SPO.

R2 - En général un programme récursif est difficile & mettre au point, et souvent pluslonga
I'exécution. Cependant pour certains problemes fondamentalement récursifs, c’est-a-dire
manipulant des structures de données récursives (arbres), la programmation récursive est
bien agréable.

SP4 : APPELS IMBRIQUES.

1) Probléme précis :

Ecrire un sous-programme P qui se contente d’afficher la valeur du parameétre A$. Ecrire
un sous-programme Q qui affecte une valeur a une variable locale A$, I'affiche et lance
I'exécution de P. Enfin, écrire un programme principal qui affecte une valeur & une varia-
ble globale A$ I'affiche et lance successivement I'exécution de P puis Q. La question qui se
pose est de savoir ce que va imprimer P lorsqu'’il est appelé depuis Q : va-t'il imprimer la
valeur de la variable locale A$ de Q ou la valeur de la variable globale A$ ?.

2) Liste du programme :

10 A$ = "PRINCIPAL”
20 PRINT A$

30CALL FN”P”

40 CALL FN "Q”(BID)
50 END

1000 DEF FN "P”

1010 PRINT "JE SUIS DANSP ... A$=";A$

1020 END FN
2000 DEF FN "Q”(BID/A$)
2010 A = ”QQQQQAQQQQ”

2020 PRINT "JE SUIS DANS Q ... A$=";A$

2030 CALL FN"P”
2040 END FN

3) Remarques :

R1 - BID est une variable "Bidon” : le sous-programme Q utilisant une variable locale doit

avoir au moins un parametre.

R2 - Un traitement analogue avec le PASCAL de I'Apple aurait donné PRINCIPAL

comme troisiéme impression.

SP5 : SOUS-PROGRAMMES PARAMETRES.

1) Probléme précis :

Ecrire un sous-programme qui calcule lasomme Z = F(A) + F(B) ot A et B sont deux fonc-
tions lues au clavier et oil F est également le nom d’une fonction définie par un sous-pro-
gramme. Ici, F pourra étre soit SINUS, soit COSINUS.

2) Liste du programme :

100REM SP5

110 PRINT "NOM DE LA FONCTION ( SINUS OU COSINUS )”
120 INPUT F$

130 PRINT "ENTREZ LA VALEUR DE A”
140 INPUT A

150 PRINT "ENTREZ LA VALEUR DE B”
160 INPUT B

170 CALL FN "CALCUL”(Z = A,B,F$)
180 PRINT "RESULTAT = "Z

190 END

200 DEF FN "CALCUL"(R = X)Y,V$/T)
210CALL FN " +V$ + ""(RX)

220 CALL FN ™"+ V$ + "(T,Y)
230R=R+T

240END FN

250 DEF FN "SINUS”(S = P)

260S = SIN (P)

270END FN

280 DEF FN ”COSINUS”(C = P)

290 C = COS (P)

300 END FN

3) Remarques :

R1 - Notez bien la syntaxe des instructions 210 et 220 dans lesquelles on lance I'exécution
de la fonction V$ dont le nom est donné dans le programme principal.

R2 - Ce type de programmation est indispensable si on veut par exemple écrire un sous-

programme général, de calcul de l'intégrale d’'une fonction quelconque (ce qui est
impossible en PASCAPPLE).

Chapitre V

EXEMPLE RECAPITULATIF

1 INTRODUCTION

2 STRUCTURE GENERALE DES DONNEES ET DES TRAITEMENTS

2 - 1 Structure des données

2 - 2 Structure des traitements

2 - 3 Le “gros” programme de gestion de cette bibliotheque
3 MODE D’UTILISATION

4 DESCRIPTION DES DIFFERENTS MODULES

1 INTRODUCTION

Le but de ce chapitre est de montrer, sur un petit exemple récapitulatif, I'utilisation
conjointe des fichiers, des masques et des sous-programmes. [l montre comment on peut
gérer les préts dans une bibliotheque, avec la simplification qu'un emprunteur ne peut
sortir qu'un livre a la fois. La structure des données et des traitements qui a été adoptée
n’est pas la seule, ni peut-étre la meilleure ; elle est exposée sans justification dans le
deuxiéme paragraphe. Le troisieme décrit schématiquement le mode d'utilisation ; vous
8tes invité a I'essayer sur votre ordinateur avant de lire le quatrieme paragraphe dan

lequel les différents modules sont expliqués et commentés. .

Bien comprendre une application de cefte envergure ne peut se faire en quelques
minutes : ne vous étonnez donc pas si I'étude de ce chapitre vous prend du temps, c'est
tout-a-fait normal. De plus, si un lecteur courageux entreprend de réécrire cette applica-
tion, pour s’ entrainer, ou une autre de difficulté comparable, qu’il ne s’étonne pas il lui
faut quelques heures, (voire quelques dizaines d’heures) pour la mettre au point. (
MEM,/DOS facilite le travail du programmeur et lui permet d’aboutir plus rapidement & un
programme qui tourne, mais il n’annule pas totalement le travail & fournir !

2 STRUCTURE GENERALE DES DONNEES ET DES TRAITEMENTS

Pour la gestion de cette bibliotheque dans laquelle les emprunteurs ne peuvent emprunter
qu’un livre & la fois, on a choisi de n’utiliser que deux fichiers et de découper le traitement
‘en plusieurs modules ayant chacun une fonction précise.

2 - 1 Structure des données

Deux fichiers indexés, un pour les livres et un pour les emprunteurs, les renseignements sur
les préts étant reportés dans ces fichiers.

Fichier LIVRES :

La clé est composée d’un code de rayon sur deux caracteres et d un numero d’ordre dans
ce rayon qui est un entier ; les rubriques sont 'auteur, le titre, I'éditeur, la date d’achat etla
date d’emprunt si ce livre est sorti.

Fichier EMPRUNTEURS : (
La clé est composée d’'un nom d’emprunteur sur 15 caractéres ; les rubriques sont d’une
part le nom suivi du prénom, I'adresse en trois zones et le numeéro de téléephone, quiidenti-
fient I'emprunteur ; d’autre part les caractéristiques du livre éventuellement emprunté :
code de rayon et numéro.

2 - 2 Structure des traitements

La mise en ceuvre de cette application nécessite trois opérations distinctes :

- une opération faite une fois pour toutes en conversationnel : créer les masques et écrire
les programmes, ce qui est le travail de I'informaticien

- une opération faite également une fois pour toutes, mais par programme: créerles fichiers

- ensuite le bibliothécaire pourra utiliser en permanence le programme général de gestion (
de sa bibliotheque.

Les masques sont décrits plus loin et leur création ne sera méme pas évoquée. Le pro-
gramme de création des fichiers est tellement simple qu’on va le liquider tout de suite - il se
compose des cing instructions ci-dessous qui ne nécessitent aucun commentaire.

100 LET "#C,$”

110 LET ")RA$2,NU%=AU$,TI$,ED$,DA$* EM$, DE$*”
120 LET " #N,F,F LIVRE”

130 LET ")EM$15=N0$,A1$,A2$,A3$,TL$, RA$ NU%"
140 LET " #N,G,F,CLIENT”

2 - 3 Le “gros” programme de gestion de cette bibliothéque

Il'y a quatre entités a gérer : un livre, un emprunteur, un prét et une restitution. Gérer un
livre ou un emprunteur nécessitent les mémes opérations : créer, lire, modifier ou détruire
un enregistrement, parcourir le fichier en séquence, et le réorganiser de temps en temps
pour diminuer les temps de recherche. Par contre, les opérations de prét et de restitution
sont trés différentes et il y aura donc deux modules distincts. Remarquons enfin que pour
des motifs de simplicité d' utilisation, il a été créé un petit module ATTENTE qui a pour seul
objet de lire une réponse au clavier. Tous ces modules sont décrits au paragraphe 4.

Le programme principal est simplement un menu. La structure des appels des modules est
décrite dans le schéma ci-dessous.

PROGRAMME
PRINCIPAL
GESTION PRET RESTITUTION
/ |
CREER MODIFIER DETRUIRE SUIVANT! REORG
\ LIRE
P. EMPRUNT P. LIVRE ATTENTE

